

# Principles of Programming Languages

## Notes

v0.10 – April 27, 2020



# Contents

|                                      |           |
|--------------------------------------|-----------|
| <b>Preface</b>                       | <b>5</b>  |
| <b>1 A Calculator</b>                | <b>7</b>  |
| 1.1 Syntax                           | 7         |
| 1.1.1 Concrete Syntax                | 7         |
| 1.1.2 Abstract Syntax                | 7         |
| 1.2 Interpreter                      | 7         |
| 1.2.1 Specification                  | 8         |
| 1.2.2 Implementation                 | 10        |
| <b>2 Simple Functional Languages</b> | <b>15</b> |
| 2.1 LET                              | 15        |
| 2.1.1 Concrete Syntax                | 15        |
| 2.1.2 Abstract Syntax                | 15        |
| 2.1.3 Environments                   | 16        |
| 2.1.4 Interpreter                    | 16        |
| 2.1.5 Inspecting the Environment     | 26        |
| 2.2 Exercises                        | 27        |
| 2.3 PROC                             | 31        |
| 2.3.1 Concrete Syntax                | 31        |
| 2.3.2 Abstract Syntax                | 31        |
| 2.3.3 Interpreter                    | 31        |
| 2.3.4 Dynamic Scoping                | 34        |
| 2.4 Exercises                        | 35        |
| 2.5 REC                              | 36        |
| 2.5.1 Concrete Syntax                | 37        |
| 2.5.2 Abstract Syntax                | 37        |
| 2.5.3 Interpreter                    | 38        |
| <b>3 Imperative Programming</b>      | <b>43</b> |
| 3.1 Mutable Data Structures in OCaml | 43        |
| 3.1.1 A counter object               | 43        |
| 3.1.2 A stack object                 | 43        |
| 3.2 EXPLICIT-REFS                    | 44        |
| 3.2.1 Concrete Syntax                | 44        |

|          |                                                 |           |
|----------|-------------------------------------------------|-----------|
| 3.2.2    | Abstract Syntax . . . . .                       | 45        |
| 3.2.3    | Interpreter . . . . .                           | 45        |
| 3.3      | IMPLICIT-REFS . . . . .                         | 49        |
| 3.3.1    | Concrete Syntax . . . . .                       | 49        |
| 3.3.2    | Abstract Syntax . . . . .                       | 50        |
| 3.3.3    | Interpreter . . . . .                           | 50        |
| 3.4      | Parameter Passing Methods . . . . .             | 54        |
| 3.4.1    | Call-by-Value . . . . .                         | 54        |
| 3.4.2    | Call-by-Reference . . . . .                     | 54        |
| 3.4.3    | Call-by-Name . . . . .                          | 54        |
| 3.4.4    | Call-by-Need . . . . .                          | 54        |
| 3.5      | Exercises . . . . .                             | 54        |
| <b>4</b> | <b>Types</b> . . . . .                          | <b>57</b> |
| 4.1      | CHECKED . . . . .                               | 57        |
| 4.1.1    | Concrete Syntax . . . . .                       | 57        |
| 4.1.2    | Abstract Syntax . . . . .                       | 57        |
| 4.1.3    | Type-Checker . . . . .                          | 58        |
| 4.1.4    | Adding Letrec . . . . .                         | 62        |
| 4.1.5    | Exercises . . . . .                             | 62        |
| <b>5</b> | <b>Modules</b> . . . . .                        | <b>65</b> |
| 5.1      | Syntax . . . . .                                | 65        |
| 5.1.1    | Concrete Syntax . . . . .                       | 66        |
| 5.1.2    | Abstract Syntax . . . . .                       | 66        |
| 5.2      | Interpreter . . . . .                           | 67        |
| 5.2.1    | Specification . . . . .                         | 67        |
| 5.2.2    | Implementation . . . . .                        | 67        |
| 5.3      | Type-Checking . . . . .                         | 68        |
| 5.3.1    | Specification . . . . .                         | 68        |
| 5.3.2    | Implementation . . . . .                        | 69        |
| 5.4      | Further Reading . . . . .                       | 70        |
| <b>A</b> | <b>Supporting Files</b> . . . . .               | <b>71</b> |
| A.1      | File Structure . . . . .                        | 71        |
| <b>B</b> | <b>Solution to Selected Exercises</b> . . . . . | <b>73</b> |

# Preface



# Chapter 1

## A Calculator

We introduce a toy language called ARITH. In ARITH programs are simple arithmetic expressions. The objective of this chapter is to provide a gentle introduction to various concepts we will be developing later in these notes. These concepts include the syntax of a language (concrete syntax, abstract syntax, parsing) and evaluation/execution of programs in a language (interpreter, specifying an interpreter, implementing an interpreter, handling errors).

### 1.1 Syntax

#### 1.1.1 Concrete Syntax

For the sake of simplicity, our language ARITH only supports two arithmetic operations, subtraction and division. We will later add others.

$$\begin{aligned}\langle \text{Expression} \rangle &::= \langle \text{Number} \rangle \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle - \langle \text{Expression} \rangle \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle / \langle \text{Expression} \rangle \\ \langle \text{Expression} \rangle &::= ( \langle \text{Expression} \rangle )\end{aligned}$$

#### 1.1.2 Abstract Syntax

```
1 type expr =  
  | Int of int  
3  | Sub of expr*expr  
  | Div of expr*expr
```

ast.ml

### 1.2 Interpreter

An interpreter<sup>1</sup> is a process that, given an expression, produces the result of its evaluation. The implementation of interpreters in these notes will be developed in two steps, first we specify the

---

<sup>1</sup>We will use the words “interpreter” and “evaluator” interchangeably.

interpreter and then we implement it proper.

- Specification of the interpreter. This includes a precise description of the possible results which a program can evaluate to, evaluation judgements that state what result a program evaluates to and a set of evaluation rules that define the meaning of evaluation judgements by describing the behavior of each construct in the language.
- Implementation of the interpreter. Using the evaluation rules of the specification as a guide, an implementation is presented. The time invested in producing the evaluation rules in the previous step, betters our understanding of the interpreter's behavior and hence diminishes the chances of introducing errors when it is implemented.

To illustrate this approach, we next specify an interpreter for ARITH and then implement it.

### 1.2.1 Specification

We begin by stating the possible results that can arise out of the evaluation of programs in ARITH. For now, we fix the set of **results** to be the integers  $\mathbb{Z}$  since evaluation of ARITH programs produce integers:

$$\mathbb{R} := \mathbb{Z}$$

The “:=” symbol is used for definitional equality, meaning here that  $\mathbb{R}$  is “defined to be” the set  $\mathbb{Z}$ . We continue with the specification of the interpreter for ARITH by introducing evaluation judgements. **Evaluation judgements** are expressions of the form:

$$e \Downarrow n$$

where  $e$  is an expression in ARITH in abstract syntax and  $n$  is a result (*i.e.*  $n$  is an integer). It is read as, “expression  $e$  evaluates to the integer  $n$ ”. The meaning of  $e \Downarrow n$  is established via so called **evaluation rules**. The preliminary set of evaluation rules of ARITH are as follows:

$$\frac{}{\text{Int}(n) \Downarrow n} \text{EInt} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m - n}{\text{Sub}(e1, e2) \Downarrow p} \text{ESub} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m / n}{\text{Div}(e1, e2) \Downarrow p} \text{EDiv}$$

Judgements above the horizontal line in an evaluation rule are called **hypothesis** and the one below is called the **conclusion**. Judgements in the hypothesis of an evaluation rule are read from left to right. In particular, note that evaluation of the arguments of all arithmetic operations proceeds from left to right. It could have been stated in the opposite order from right-to-left and, at this point in time, does not make much of a difference<sup>2</sup>. A **derivation tree** is a tree of evaluation judgements whose nodes are instances of evaluation rules. A judgement  $e \Downarrow n$  is **derivable or holds** if there is a derivation tree with  $e \Downarrow n$  as its root.

<sup>2</sup>But later in our development, when evaluation of expressions can cause certain effects (such as modifying mutable data structures), this difference will become relevant.



$$\begin{array}{c}
\frac{}{\text{Int}(n) \Downarrow n} \text{EInt} \\
\\
\frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m - n}{\text{Sub}(e1, e2) \Downarrow p} \text{ESub} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m/n}{\text{Div}(e1, e2) \Downarrow p} \text{EDiv} \\
\\
\frac{e1 \Downarrow \text{error}}{\text{Sub}(e1, e2) \Downarrow \text{error}} \text{ESubErr1} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow \text{error}}{\text{Sub}(e1, e2) \Downarrow \text{error}} \text{ESubErr2} \\
\\
\frac{e1 \Downarrow \text{error}}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr1} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow \text{error}}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr2} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow 0}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr3}
\end{array}$$

---

**Figure 1.1:** Evaluation rules for ARITH

**Example 1.2.1.** Two examples of derivable evaluation judgements are  $\text{Sub}(\text{Int } 3, \text{Int } 1) \Downarrow 2$  and also  $\text{Sub}(\text{Div}(\text{Int } 8, \text{Int } 2), \text{Int } 1) \Downarrow 3$ .

We are not quite done with the task of specifying our interpreter since not all expressions in ARITH return numbers. For example,  $\text{Div}(\text{Int } 2, \text{Int } 0)$  does not evaluate to any number. Rather it should evaluate to an error. Thus, the above mentioned evaluation rules are incomplete since there are situations that are left unspecified. It is important to have a complete set of rules so that when we implement the interpreter there are no ambiguities. Moreover, we must revisit our notion of result since it should include an error as a possible outcome. The previously introduced evaluation judgements are thus revisited below. The final form that evaluation judgements take for ARITH are:

$$e \Downarrow r$$

where  $r$  denotes a result of the computation  $r \in \mathbb{R}$ . A result is either an integer or a special element *error*. In other words, we set  $\mathbb{R} := \mathbb{Z} \cup \{\text{error}\}$ . The subset of results that are integers are called expressed values: it is the name given to non-error results of evaluation.

The evaluation rules defining this new judgement, and therefore the evaluation rules for ARITH, are those presented in Fig. 1.1, where  $m, n, p \in \mathbb{Z}$ . The first three rules are the ones already presented above. Rules ESubErr1, ESubErr2, EDivErr1, and EDivErr2 state how error propagation takes place. The last one introduces a new error, namely division by zero. Moving forward, and for the sake of brevity, we will not be specifying the error propagation rules when specifying the interpreter. We will only be presenting the error introduction rules.

We next address the implementation of an interpreter for ARITH. We will do so in two attempts, first a preliminary attempt and then a final one. The preliminary attempt has various drawbacks that we will point out along the way but has the virtue of serving as a convenient stepping stone towards the final one.



A summary of some important concepts we have covered are:

- Result
- Expressed Value
- Evaluation Judgement
- Evaluation Rules
- Derivation Tree
- Derivable Evaluation Judgements

## 1.2.2 Implementation

In order to use the evaluation rules as a guideline for our implementation we first need to model both components of evaluation judgements in OCaml, namely expression  $e$  and result  $r$  in  $e \Downarrow r$ . The former is already expressed in abstract syntax, which we encoded as the algebraic data type `expr` in OCaml. So that item has already been addressed. As for the latter, since it denotes either an integer or an error, we will model it in OCaml using the following type<sup>3</sup>:

```
type 'a result = Ok of 'a | Error of string
```

For example, the type `int result` may be read as a type that states that “the result of the evaluator is an integer or an error”. Likewise, `bool result` may be read as a type that states that “the result of the evaluator is a boolean or an error”. A result can either be a meaningful value of type `'a` prefixed with the constructor `Ok`, or else an error with an argument of type `string` prefixed with an `Error` constructor. For example, `Ok 3` has type `int result`.



In a type expression such as `int result`, we say `int` is a “type” and `int result` is a “type”. But we refer to `result` as a **type constructor** since, given a type `'a`, it constructs a type `'a result`.

We can now proceed with an implementation of an interpreter for ARITH following the evaluation rules as close as possible. If we call our evaluator function `eval_expr`, its type can be expressed as follows, indicating that evaluation consumes an expression and returns either an integer or an error with a string description:

```
eval_expr : exp -> int result
```

The code is given in Fig. 1.2. This figure is an example of a code listing. We occasionally add an indication to code listings as to where the snippet of code resides. For example, in this case it resides in file `interp.ml`. Notice that `eval_expr` is a recursive function over the structure of expressions in abstract syntax (*i.e.* values of type `expr`). In the first clause, `Int(n)`, it simply returns `Ok n`. Note that returning just `n` would be incorrect since our interpreter must return a value of type `int result`, not of type `int`. In the clause for `Sub(e1,e2)`, the `match` keyword forces evaluation of `e1` before `e2`, as indicated by the evaluation rules<sup>4</sup>. A similar comment applies to the other arithmetic operation. One notices, too, that a substantial amount of code

<sup>3</sup>OCaml has a built-in type `type ('a,'b) result = Ok of 'a | Error of 'b`. We could have used this type but it is slightly more general than necessary since our errors will always be accompanied by a string argument rather than different types of arguments.

<sup>4</sup>OCaml evaluates arguments from right to left.

```

1 let rec eval_expr : expr -> int result =
2   fun e ->
3     match e with
4     | Int(n) -> Ok n
5     | Sub(e1,e2) ->
6       (match eval_expr e1 with
7        | Error s -> Error s
8        | Ok m -> (match eval_expr e2 with
9                   | Error s -> Error s
10                  | Ok n -> Ok (m-n)))
11
12    | Div(e1,e2) ->
13      (match eval_expr e1 with
14       | Error s -> Error s
15       | Ok m -> (match eval_expr e2 with
16                  | Error s -> Error s
17                  | Ok n -> if n==0
18                           then Error "Division by zero"
19                           else Ok (m/n)))

```

interp.ml

**Figure 1.2:** Preliminary Interpreter for ARITH

checks for errors and then propagates them.. Notice too that, in the `Div` case, in addition to propagating errors resulting from evaluating its arguments `e1` and `e2`, it generates a new one if the denominator is 0. The only error modeled in ARITH is division by zero. An example of error propagation takes place in an ARITH expression such a `Add(Div(Int 1,Int 0),e)`, where `e` can be any expression. Here the expression `e` is never evaluated since evaluation of `Div(Int 1,Int 0)` produces `Error "Division by zero"`, hence `e` is ignored and `Error "Division by zero"` is immediately produced as the final result of evaluation of the entire expression.

Although certainly necessary, there is no interesting computational content in error propagation. It would be best to have it be handled behind the scenes, by appropriate error propagation helper functions.

### Implementation: Final

Following our comment on error propagation, we introduce three helper functions for this purpose: `return`, `error` and `>>=` (pronounced “bind”). The code for these functions is given in Fig. 1.3. The `return` function simply returns its argument inside an `Ok` constructor and may thus be seen as producing a non-error result. A similar comment applies to `error`: given a string it produces an error result by simply prefixing the string with the `Error` constructor. The infix operator `>>=` is called `bind` and is left associative<sup>5</sup>. It takes an expression `c` that returns a result (i.e. a non-error value or an error value), and if `c` returns an error, propagates it. Otherwise, if it evaluates to `Ok v`, for some `v`, then it passes `v` on to `f` by evaluating `f v`.

An alternative description of these helper functions is as follows. Let us dub expressions of type `int result`, **structured programs**. We could have taken the more general type `'a result`, rather than `int result`, as our notion of structured programs, but the latter will suffice for our explanation. Structured programs may be seen as are programs that, apart from producing

<sup>5</sup>The precedence and associativity of user-defined infix/prefix operators may be consulted here: <https://caml.inria.fr/pub/docs/manual-caml-light/node4.9.html>

```

let return : 'a -> 'a result =
2   fun v -> Ok v

let error : string -> 'a result =
4   fun s -> Error s

6   let (>=) : 'a result -> ('a -> 'b result) -> 'b result =
8   fun c f ->
    match c with
10  | Error s -> Error s
    | Ok v -> f v

```

ds.ml

Figure 1.3: The Error Monad

an integer as end product, can manipulate additional structure such as error handling, state, non-determinism, etc. In our particular case, a structured program handles errors as additional structure. Under this light, we can describe the helper functions as follows:

- `return` may be seen as a function that creates a (trivial) structured program that returns an integer (*i.e.* non-error) result.
- `error` may be seen as a function that creates a (trivial) structured program that returns an error result.
- `>=` may be seen as a function that models composition of a structured program `c` and a function `f` that, given an integer, produces a structured program. Allowing `f >=` to be a function that produces a structured program, rather than just a structured program, allows `f` to use the result of evaluating the first argument.

Let us rewrite our interpreter for our simple expression language using these helper functions.

```

let rec eval_expr : expr -> int result =
2   fun e ->
    match e with
4   | Int(n) -> return n
    | Sub(e1,e2) ->
6       eval_expr e1 >= (fun n1 ->
7         eval_expr e2 >= (fun n2 ->
8           return (n1-n2)))
    | Div(e1,e2) ->
10      eval_expr e1 >= (fun n1 ->
11        eval_expr e2 >= (fun n2 ->
12          if n2==0
13            then error "Division by zero"
14            else return (n1/n2)))

```

Consider the code for the `Sub(e1,e2)` case. Notice how if `eval_expr e1` produces an error result, say `Error "Division by zero"` because `e1` had a division by zero, then `>=` simply ignores its second argument, namely the expression `(fun n1 -> eval_expr e2 >= (fun n2 -> return (n1+n2)))`, and returns the error result `Error "Division by zero"` immediately as the final result of the evaluation, thus effectively propagating the error.

In fact, we can further simplify this code by dropping superfluous parenthesis. This leads to our final evaluator for ARITH expressions.

```

let rec eval_expr : expr -> int result =
2  fun e ->
  match e with
4  | Int(n) -> return n
  | Sub(e1,e2) ->
6    eval_expr e1 >>= fun n1 ->
    eval_expr e2 >>= fun n2 ->
8    return (n1-n2)
  | Div(e1,e2) ->
10   eval_expr e1 >>= fun n1 ->
    eval_expr e2 >>= fun n2 ->
12   if n2==0
    then error "Division by zero"
14   else return (n1/n2)

```

Some additional observations on the behavior of the error handling operations:

$$\begin{aligned}
 \text{return } e >>= f &\simeq f\ e \\
 m >>= \text{return} &\simeq m \\
 (m >>= f) >>= g &\simeq m >>= (\text{fun } x \rightarrow f\ x >>= g) \\
 \text{error } e >>= f &\simeq \text{error } e
 \end{aligned}$$

The symbol  $\simeq$  above means that the left and right hand sides of these equations behave the same way.



The `result` type, together with the operations `return`, `error` and `>>=` is called an **Error Monad**. Monads are well-known in pure functional programming languages like Haskell, where they allow to handle side-effects behind the scenes without compromising equational reasoning (see the equations presented above). However, they are also important in non-pure functional languages, like OCaml, where they are a means to better structure one's code, as we have seen from our use of it here.



## Chapter 2

# Simple Functional Languages

## 2.1 LET

### 2.1.1 Concrete Syntax

```
⟨Program⟩    ::=  ⟨Expression⟩

⟨Expression⟩ ::=  ⟨Number⟩
⟨Expression⟩ ::=  ⟨Identifier⟩
⟨Expression⟩ ::=  ⟨Expression⟩⟨BOp⟩⟨Expression⟩
⟨Expression⟩ ::=  zero?(⟨Expression⟩)
⟨Expression⟩ ::=  if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::=  let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::=  (⟨Expression⟩)

⟨BOp⟩        ::=  + | - | * | /
```

### 2.1.2 Abstract Syntax

```
2  type prog = AProg of expr
4  type expr =
6    | Int of int
6    | Var of string
8    | Add of expr*expr
8    | Sub of expr*expr
10   | Mul of expr*expr
10   | Div of expr*expr
12   | IsZero of expr
12   | ITE of expr*expr*expr
12   | Let of string*expr*expr
```

### 2.1.3 Environments

Consider the LET expression  $x+2$ . We cannot determine the result of its evaluation because, in a sense, it is incomplete. Indeed, unless we are given the value assigned to the identifier  $x$ , we cannot determine whether evaluation of  $x+2$  should result in an error (if say,  $x$  held the value  $\text{true}$ <sup>1</sup>), or a number such as 4 (if, say,  $x$  held the value 2). Therefore, evaluation of expressions in LET require an assignment of values to identifiers. These assignments are called **environments**. The interpreters developed in these notes are therefore known as **environment-based interpreters** as opposed to **substitution-based interpreters**, in the latter values are substituted directly into the expressions rather than recording them (and then looking them up) in environments.

An **environment** is a partial function from the set of identifiers to the set of expressed values. **Expressed values**, denoted  $\mathbb{EV}$ , are the set of values that are not errors that we can get from evaluating expressions. In ARITH the only expressed values are the integers. In LET they are the integers and the booleans:

$$\mathbb{EV} := \mathbb{Z} \cup \mathbb{B}$$

If  $\mathbb{ID}$  denotes the set of all identifiers, then we can define the set of all environments  $\mathbb{ENV}$  as follows.

$$\mathbb{ENV} := \mathbb{ID} \rightarrow \mathbb{EV}$$

We use letters  $\rho$  and  $\rho'$  to denote environments. For example,  $\rho = \{x := 1, y := 2, z := \text{true}\}$  is an environment that assigns 1 to  $x$ , 2 to  $y$  and  $\text{true}$  to  $z$ . We write  $\rho(id)$  for the value associated to the identifier  $id$ . For example,  $\rho(x)$  is 1.

### 2.1.4 Interpreter

#### Specification

As you might recall from our presentation of ARITH, evaluation of an ARITH expression produces a result. A result could either be an integer or an error. We can still get an error from evaluating a LET expression since ARITH expressions are included in LET expressions. However, if there is no error, then in LET we can either get an integer or a boolean. The set of results for LET is thus:

$$\mathbb{R} := \mathbb{EV} \cup \{\text{error}\}$$

where  $\mathbb{EV}$  was updated above during our discussion on environments. Evaluation judgements for LET include an environment:

$$e, \rho \Downarrow r$$

It should be read as follows, “evaluation of expression  $e$  under environment  $\rho$ , produces result  $r$ ”. The rules defining this judgement are presented in Fig. 2.1. Rule EVar performs lookup in the current environment. The related rule EVarErr models the error resulting from lookup failing to find the identifier in the environment. The rules for addition, subtraction and multiplication are similar as the one for division and omitted. The notation  $\rho \oplus \{id := w\}$  used in ELet stands for the environment that maps expressed value  $w$  to identifier  $id$  and behaves as  $\rho$  on all other identifiers.

---

<sup>1</sup>We do not have booleans in ARITH but we will in LET.



$$\begin{array}{c}
\frac{}{\text{Int}(n), \rho \Downarrow n} \text{EInt} \quad \frac{\rho(\text{id}) = v}{\text{Var}(\text{id}), \rho \Downarrow v} \text{EVar} \\
\\
\frac{\text{e1} \Downarrow m \quad \text{e2}, \rho \Downarrow n \quad p = m/n}{\text{Div}(\text{e1}, \text{e2}), \rho \Downarrow p} \text{EDiv} \\
\\
\frac{\text{e}, \rho \Downarrow v \quad v = 0}{\text{IsZero}(\text{e}), \rho \Downarrow \text{true}} \text{EIZTrue} \quad \frac{\text{e}, \rho \Downarrow v \quad v \neq 0}{\text{IsZero}(\text{e}), \rho \Downarrow \text{false}} \text{EIZFalse} \\
\\
\frac{\text{e1}, \rho \Downarrow \text{true} \quad \text{e2}, \rho \Downarrow v}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow v} \text{EITETrue} \quad \frac{\text{e1}, \rho \Downarrow \text{false} \quad \text{e3}, \rho \Downarrow v}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow v} \text{EITEFalse} \\
\\
\frac{\text{e1}, \rho \Downarrow w \quad \text{e2}, \rho \oplus \{\text{id} := w\} \Downarrow v}{\text{Let}(\text{id}, \text{e1}, \text{e2}), \rho \Downarrow v} \text{ELet} \\
\\
\frac{\text{id} \notin \text{dom}(\rho)}{\text{Var}(\text{id}), \rho \Downarrow \text{error}} \text{EVarErr} \quad \frac{\text{e1}, \rho \Downarrow m \quad \text{e2}, \rho \Downarrow 0}{\text{Div}(\text{e1}, \text{e2}), \rho \Downarrow \text{error}} \text{EDivErr} \\
\\
\frac{\text{e}, \rho \Downarrow v \quad v \notin \mathbb{Z}}{\text{IsZero}(\text{e}), \rho \Downarrow \text{error}} \text{EIZErr} \quad \frac{\text{e1}, \rho \Downarrow v \quad v \notin \mathbb{B}}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow \text{error}} \text{EITEErr}
\end{array}$$

---

**Figure 2.1:** Evaluation Semantics for LET

**Implementation: Attempt I**

Before implementing the evaluator we must first implement expressed values and environments. Expressed values can be naturally described using algebraic data types. Environments can be modeled in various ways in OCaml: as functions, as association lists, as hash tables, as algebraic data types, etc. Due to its simplicity we follow the last of these.

```

type exp_val =
2   | NumVal of int
   | BoolVal of bool
4
type env =
6   | EmptyEnv
   | ExtendEnv of string*exp_val*env

```

ds.ml

Operations on environments are:

```

1  let empty_env : unit -> env =
   fun () -> EmptyEnv
3
let extend_env : env -> string -> exp_val -> env =
5   fun env id v -> ExtendEnv(id,v,env)
7
let rec apply_env : string -> env -> exp_val result =
   fun id env ->
9   match env with
   | EmptyEnv -> error @@ id^" not found!"
11  | ExtendEnv(v,ev,tail) ->
   if id=v
13   then return ev
   else apply_env id tail

```

ds.ml



The @@ operator is right-associative function application. It helps reduce the number of required parenthesis. For example, given a function `succ: int -> int` that returns the successor of an integer, the expression `succ @@ succ @@ succ @@ succ 1` is equivalent to `succ (succ (succ (succ 1)))`.

Notice that `apply_env en id` has `exp_val result` as return type because it returns an error if `id` is not found in the environment `en`. However, if there is an expressed value associated to `id` in `en`, then that will be returned (wrapped with an `Ok` constructor).

Next we implement an interpreter for LET by following the evaluation rules of Fig. 2.1 closely. Indeed, the evaluation rules shall serve as a specification for, and thus guide, our implementation. The code itself is given in Fig. 2.2. The case for `Var(id)` simply invokes `apply_env` to look up the expressed value associated to the identifier `id` in the environment `env`. The case for `Div(e1,e2)` makes use of an auxiliary operation `int_of_numVal`, explained below.

```

let int_of_numVal : exp_val -> int result = function
2   | NumVal n -> return n
   | _ -> error "Expected a number!"

```

ds.ml

```

let rec eval_expr : expr -> env -> exp_val result =
2   fun e en ->
    match e with
4   | Int(n) -> return @@ NumVal n
    | Var(id) -> apply_env id en
6   | Div(e1,e2) -> (* Add, Sub and Mul are similar and omitted *)
        eval_expr e1 en >=>
8       int_of_numVal >=> fun n1 ->
        eval_expr e2 en >=>
10      int_of_numVal >=> fun n2 ->
        if n2==0
12      then error "Division by zero"
        else return @@ NumVal (n1/n2)
14  | IsZero(e) ->
        eval_expr e en >=>
16      int_of_numVal >=> fun n ->
        return @@ BoolVal (n = 0)
18  | ITE(e1,e2,e3) ->
        eval_expr e1 en >=>
20      bool_of_boolVal >=> fun b ->
        if b
22      then eval_expr e2 en
        else eval_expr e3 en
24  | Let(id,def,body) ->
        eval_expr def en >=> fun ev ->
26      eval_expr body (extend_env en id ev)

```

interp.ml

Figure 2.2: Preliminary Interpreter for LET

Evaluation of  $e_1$  in  $\text{Div}(e_1, e_2)$  could produce an expressed value other than a number (*i.e.* other than a `NumVal`). The function `int_of_numVal` checks to see whether its argument is a `NumVal` or not, returning a result that consists of an error, if it is not, or else the number itself (without the `NumVal` tag). This number is then bound to the variable `n1`. A similar description determines the value of `n2`. Finally, if `n2` is zero, an error is returned, otherwise the desired quotient is produced as a result.

The code for the other binary operators, for the zero predicate and for the conditional are similar. The case for `ITE` uses a similar helper function `bool_of_boolVal`. The last case, `Let`, evaluates the definition and then extends the environment appropriately before evaluating the body. Notice how local scoping is implemented by adding a new entry into the environment.

The top level function for the interpreter is called `interp`. It parses the string argument, evaluates producing a function `c` that awaits an environment, and then feeds that function the empty environment `EmptyEnv`.

```

let interp (s:string) : exp_val result =
2   let c = s |> parse |> eval_prog
    in c EmptyEnv

```

interp.ml

Here is an example run of our interpreter<sup>2</sup>:

<sup>2</sup>Negative numbers must be placed between parenthesis. For example, `interp "(-7)"` rather than `interp "-7"`.

```

1 utop # interp "
2 let x=2
3 in let y=3
4 in x+y";;
- : exp_val Ds.result = Ok (NumVal 5)

6
7 utop # interp "
8 let x=2
9 in let y=0
10 in x+(x/y)";;
- : exp_val Ds.result = Error "Division by zero"

```

utop



In ARITH there was only one possible error that could be introduced and propagated, namely the division by zero error. In LET there are four possible errors that can be introduced and propagated: division by zero, identifier not found, expected a number and expected a boolean.

### Weaving Environments

Our code for LET seems well-structured and robust enough to be extensible to additional language features. Even so, we can perhaps take a step further. Notice that the environment is explicitly threaded around the entire program. Consider the following excerpt from Fig. 2.2 and notice how the environment (highlighted) is passed on to each occurrence of `eval_expr`.

```

1 let rec eval_expr : expr -> env -> exp_val result =
2   fun e en ->
3     match e with
4     | ...
5     | ITE(e1,e2,e3) ->
6       eval_expr e1 >>=
7         bool_of_boolVal >>= fun b ->
8         if b
9         then eval_expr e2 en
10        else eval_expr e3 en

```

The reason `en` is passed on in each case above, is that all expressions `e1`, `e2` and `e3` are evaluated under that same environment. This occurs with other language constructs too such as `Add(e1,e2)`, `Div(e1,e2)`, `Sub(e1,e2)` and `Mul(e1,e2)`, where both `e1` and `e2` are evaluated under the environment `en`. An alternative would be to have the environment be passed around “behind the scenes”, in the same way that error propagation is handled behind the scenes. The resulting code would look something like this, where all occurrences of `en` have been removed:

```

1 let rec eval_expr : expr -> exp_val ea_result =
2   fun e en ->
3     match e with
4     | ...
5     | ITE(e1,e2,e3) ->
6       eval_expr e1 >>=
7         bool_of_boolVal >>= fun b ->
8         if b

```

10

```

    then eval_expr e2
    else eval_expr e3

```

Here `eval_expr e1` now produces `a`:

function that waits for the environment and then produces a result.

This environment will be supplied when we run the program, and then propagated by `>>=` (which we would need to update so that it continues to propagate errors but now also propagates environments). This of course requires that we update our definition of the `result` type. Rather than having our evaluator be of type

```
eval_expr : expr -> env -> exp_val result
```

we must “push” the environment argument into the result type `exp_val result` itself, so that the type of `eval_expr` becomes:

```
eval_expr : expr -> exp_val ea_result
```

Indeed, the new `ea_result` type constructor, read “environment abstracted result”, is defined by simply abstracting the type of environments over the standard result type:

```
type 'a ea_result = env -> 'a result
```

Since the argument of type `env` in `eval_expr` has now been pushed into the `ea_result` type constructor, this, in turn, will require the bind helper function `>>=` for `result` to be updated for the new `ea_result` so that it now passes around the environment for us. Also, `return` and `error` will need to be updated.



`eval_expr` still requires both an expression and an environment as arguments. Only now, when coding it, we will only ever explicitly mention the expression argument, leaving the helper functions to pass the environment around for us. This becomes clear from the final code for our interpreter in Fig. 2.4.

The definition of `ea_result` and its supporting operations are given in Fig. 2.3. Note how `>>=` now passes the environment argument `env` first to `c` and then to `f v`, thus effectively threading the environment for us. Also, we have a new operation `run` that given an environment abstracted result, will feed it the empty environment and thus perform the computation itself resulting in either an `ok` value or an `error` value. This operation is only used once, at the top-level, as may be seen from the following code:

2

```

let interp (e:string) : exp_val result =
  let c = e |> parse |> eval_prog
  in run c

```

`interp.ml`

The variable `c` is used as mnemonic for “computation” (also referred to as a “structured program”) the program that results from evaluating the abstract syntax tree of `e`. The computation is executed by passing it the empty environment.

You may safely ignore `>>+` for now, we’ll explain it later.

```
2 type 'a result = Ok of 'a | Error of string
4
6 type 'a ea_result = env -> 'a result
8
10 let return : 'a -> 'a ea_result =
12   fun v ->
14     fun env -> Ok v
16
18 let error : string -> 'a ea_result =
20   fun s ->
22     fun env -> Error s
24
26 let (>=) : 'a ea_result -> ('a -> 'b ea_result) -> 'b ea_result =
28   fun c f ->
29     fun env ->
30       match c env with
31       | Error err -> Error err
32       | Ok v -> f v env
33
34 let (>+) : env ea_result -> 'a ea_result -> 'a ea_result =
35   fun c d ->
36     fun env ->
37       match c env with
38       | Error err -> Error err
39       | Ok newenv -> d newenv
40
41 let run : 'a ea_result -> 'a result =
42   fun c -> c EmptyEnv
```

ds.ml

**Figure 2.3:** The Reader and Error Monad Combined



Compare Fig. 2.3 with Fig. 1.3. A value of the expressed result type of our interpreter for ARITH, could either be an ok value or an error value. A value of the expressed result type of our interpreter for LET on the other hand, will be a function that, given an environment, returns either an ok value or an error value.

### Implementation: Final

We next revisit our evaluator for LET, this time making use of our new environment abstracted result type. It is given in Fig. 2.4. We briefly comment on some of the variants.

The code for the `Int(n)` variant, remains unaltered:

```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   | Int(n) -> return @@ NumVal n

```

Note, however, that `return @@ NumVal n` now returns a function that given an environment, ignores it and simply returns `Ok (NumVal n)`. The `Var(id)` variant is slightly different. Now `apply_env` is applied only to the argument `id`, thus producing an expression (through partial application) that waits for the second argument, namely the environment. This environment will be supplied when we run the computation (using `run`).

The variants `Div(e1,e2)`, `IsZero(e)` and `ITE(e1,e2,e3)` are as in Fig. 2.2 except that the environment argument has been dropped. Finally, consider `Let(id,def,body)`. Let us recall from Fig. 2.2, the code we had for this variant:

```

| Let(id,def,body) ->
2   eval_expr def en >>= fun ev ->
    eval_expr body (extend_env en id ev)

```

We first evaluate `def` in the current environment `en` producing an expressed value `ev`. This expressed value is used to extend the current environment `ev`, before evaluating the body `body`. Dropping `en`, which is now threaded implicitly for us, results in:

```

| Let(id,def,body) ->
2   eval_expr def >>= fun ev ->
    eval_expr body (extend_env id ev)

```

There are two problems with this code. First we need to be able to produce the modified environment **as a result** so that we can pass it on when evaluating `body`. This is achieved by updating `extend_env`, and `empty_env` too although we will not be needing the latter for now, that produces the updated environment as a result (notice the `env` in `env ea_result`):

```

let extend_env : string -> exp_val -> env ea_result =
2   fun id v ->
    fun env -> Ok (ExtendEnv(id,v,env))

```

With this new operation we can produce:

```

1   | Let(id,def,body) ->
    eval_expr def >>= fun ev ->
3   extend_env id ev >>???
    eval_expr body

```

which can be simplified to

```

1 | Let(id,def,body) ->
2   eval_expr def >>=
3   extend_env id >>???
4   eval_expr body

```

This code evaluates `def` under the current environment threaded by `bind`, then feeds the resulting expressed value into `extend_env id` to produce an extended environment. But now we are faced with a second problem. It is this extended environment that should be fed into `eval_expr body` and **not** the current environment that is threaded by `bind` (the current environment presumably has no mapping for `id`). This suggests introducing the following environment update operation:

```

1 let (>>+) : env ea_result -> 'a ea_result -> 'a ea_result =
2   fun c d ->
3   fun env ->
4   match c env with
5   | Error err -> Error err
6   | Ok newenv -> d newenv

```

An expression such as `c >>+ d` first evaluates `c env`, where `env` is the current environment, producing an environment `newenv` as a result, and then completely ignores the current environment `env` feeding that new environment as current environment for `d`.

With the help of environment update, we can now complete our code for `Let`:

```

1 | Let(id,def,body) ->
2   eval_expr def >>=
3   extend_env id >>+
4   eval_expr body

```



You can think of `c1 (>>=) f` as a form of composition of computations, “given an environment `en`, pass it on to `c1` producing an expressed value `v`, then pass `v` and `en` on to `f`, and return its result as the overall result”. While `c1 (>>+) c2` may be thought of as, “given an environment `en`, pass it on to `c1` producing an environment `newenv` (not an expressed value!) as a result which is passed on to computation `c2`, returning the latter’s result as the result of the overall computation.”

Note the absence of all references to `en` in Listing. 2.4. Indeed, the environment will be passed around when we execute `run c`. According to the definition of `run`, `run c` just applies `c` to the empty environment `EmptyEnv`.

**Example 2.1.1.** We conclude this section with some examples of expressions whose type involve the `ea_result` type constructor:



```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
   match e with
4   | Int(n) ->
      return @@ NumVal n
6   | Var(id) ->
      apply_env id
8   | Div(e1,e2) -> (* Add, Sub and Mul are similar and omitted *)
      eval_expr e1 >>=
10     int_of_numVal >>= fun n1 ->
      eval_expr e2 >>=
12     int_of_numVal >>= fun n2 ->
      if n2==0
14     then error "Division by zero"
      else return @@ NumVal (n1/n2)
16   | IsZero(e) ->
      eval_expr e >>=
18     int_of_numVal >>= fun n ->
      return @@ BoolVal (n = 0)
20   | ITE(e1,e2,e3) ->
      eval_expr e1 >>=
22     bool_of_boolVal >>= fun b ->
      if b
24     then eval_expr e2
      else eval_expr e3
26   | Let(id,def,body) ->
      eval_expr def >>=
28     extend_env id >>+
      eval_expr body
30   | _ -> error "Not implemented yet!"
and
32   eval_prog (AProg e) = eval_expr e

34 let parse s =
   let lexbuf = Lexing.from_string s in
36   let ast = Parser.prog Lexer.read lexbuf in
   ast
38
40 let interp (e:string) : exp_val result =
   let c = e |> parse |> eval_prog
   in run c

```

interp.ml

Figure 2.4: Evaluator for LET

| Expression                             | Type                           | Informal Description                                                                                                                                                                                                                   |
|----------------------------------------|--------------------------------|----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| <code>return (NumVal 3)</code>         | <code>exp_val ea_result</code> | Denotes a function that when given an environment, ignores it, and immediately returns <code>Ok (NumVal 3)</code> .                                                                                                                    |
| <code>error "oops"</code>              | <code>'a ea_result</code>      | Denotes a function that when given an environment, ignores it, and immediately returns <code>Error "oops"</code> .                                                                                                                     |
| <code>apply_env "x"</code>             | <code>exp_val ea_result</code> | Denotes a function that when given an environment, inspects it to find the expressed value $v$ associated to <code>"x"</code> . If it finds it, it returns <code>Ok v</code> , otherwise it returns <code>Error "x not found"</code> . |
| <code>extend_env "x" (NumVal 3)</code> | <code>env ea_result</code>     | Denotes a function that when given an environment, extends it producing a new environment <code>newenv</code> , with the new key-value pair $x := \text{NumVal } 3$ , and returns <code>Ok newenv</code> .                             |

### 2.1.5 Inspecting the Environment

It is often convenient to be able to inspect the contents of the environment as a means of understanding how evaluation works or simply for debugging purposes. This section extends LET with a new expression `debug(e)` whose evaluation will print the contents of the current environment and ignore  $e$ . We first add a new production to the grammar defining the concrete syntax of LET:

$$\langle \text{Expression} \rangle ::= \text{debug}(\langle \text{Expression} \rangle)$$

We next add a new variant to the type `expr` defining the abstract syntax of LET:

```
2 type expr =
  ...
  | Debug of expr
```

The next step is to specify, and then implement, the extension to the interpreter for LET that handles the new construct. It is not obvious how to proceed. In particular, what should we choose as the value resulting from evaluating `Debug(e)`? After all, `Debug(e)` does not denote any value, rather it is expected to cause an effect, namely print the contents of the environment on the screen.

$$\frac{}{\text{Debug}(e), \rho \Downarrow ???} \text{EDebug}$$

In principle, we could choose any number or boolean as the value resulting from evaluating `Debug(e)`. However, any such choice would be ad-hoc. For the sake of clarity, we introduce a new expressed value specifically targeted as the result of evaluating expressions that are intended to cause effects rather than denote a meaningful value. This new value is called *unit*. Thus the set of expressed values becomes:

$$\text{EV} := \mathbb{Z} \cup \mathbb{B} \cup \mathbb{U}$$

where  $\mathbb{U} := \{\text{unit}\}$ . Its implementation is:

```

1 type exp_val =
  | NumVal of int
3   | BoolVal of bool
  | UnitVal

```

ds.ml

The amended evaluation rule EDebug becomes:

$$\frac{}{\text{Debug}(e), \rho \Downarrow \text{unit}} \text{EDebug}$$

Finally, the implementation of this evaluation rule is given below. It makes use of an auxiliary function `string_of_env`, defined in `ds.ml`, which traverses an environment and returns a string representation of it.

```

eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4     ...
    | Debug(e) ->
6       string_of_env >>= fun str ->
          print_endline str;
8       return @@ UnitVal

```

Note that there is a slight discrepancy between the specification of the evaluation rule describing how `Debug(e)` is evaluated (*i.e.* EDebug) and our implementation. Indeed, the latter prints two strings on the screen but the former does not mention any side-effects such as printing. The reason for this mismatch is that we have decided to keep the specification of our interpreters as simple as possible. In particular, we have decided not to model side-effects such as printing on the screen. Later we will show how to model other side-effecting operations when specifying interpreters. Notably, we will include an assignment operation in our language.

## 2.2 Exercises

**Exercise 2.2.1** ( $\diamond$ ). Write an OCaml expression of each of the types below:

1. *expr*
2. *env*
3. *exp\_val*
4. *exp\_val result*
5. *int result*
6. *env result*
7. *int ea\_result*
8. *exp\_val ea\_result*
9. *env ea\_result*

**Exercise 2.2.2.** Consider the following code:

```

open Ds
2
let c =
4   empty_env () >>+
   extend_env "x" (NumVal 1) >>+
6   extend_env "y" (BoolVal false) >>+
   string_of_env

```

where the helper function `string_of_env` is defined as follows:

```

let string_of_expval = function
2  | NumVal n -> "NumVal " ^ string_of_int n
  | BoolVal b -> "BoolVal " ^ string_of_bool b
4  | UnitVal -> "UnitVal "

6  let rec string_of_env' = function
  | EmptyEnv -> ""
8  | ExtendEnv(id,v,env) ->
   string_of_env' env ^ ("^id^", "^string_of_expval v^")\n"

10
12 let string_of_env : string ea_result =
   fun env -> Ok ("Environment:\n" ^ string_of_env' env)

```

ds.ml

1. Fill in the missing parenthesis in the definition of `c`. Recall that `>>+` associates to the left.
2. What is the type of `c`?
3. What happens when you load the code into `utop` and type `c`?
4. What happens when you load it into `utop` and type `run c`?

**Exercise 2.2.3.** Consider the following extension of LET with pairs. Its concrete syntax includes all the grammar productions of LET plus:

$$\begin{aligned}
 \langle \text{Expression} \rangle &::= \text{pair}(\langle \text{Expression} \rangle, \langle \text{Expression} \rangle) \\
 &\quad | \text{fst}(\langle \text{Expression} \rangle) \\
 &\quad | \text{snd}(\langle \text{Expression} \rangle)
 \end{aligned}$$

Examples of programs in this language are

1. `pair (2,3)`
2. `pair (pair(7,9),3)`
3. `pair(zero?(4),11-x)`
4. `snd(pair (pair(7,9),3))`

The abstract syntax includes the following additional variants:

```

type expr =
2  ...
  | Pair of expr*expr
4  | Fst of expr
  | Snd of expr

```

Specify the interpreter (i.e. its evaluation rules) and then implement it.

**Exercise 2.2.4.** Consider another extension to LET with pairs. Its concrete syntax is:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \text{pair}(\langle \text{Expression} \rangle, \langle \text{Expression} \rangle) \\ &\quad | \text{unpair}(\langle \text{Identifier} \rangle, \langle \text{Identifier} \rangle) = \langle \text{Expression} \rangle \text{ in } \langle \text{Expression} \rangle \end{aligned}$$

Pairs are constructed in the same way as in Exercise 2.2.3. However, to eliminate pairs instead of *fst* and *snd* we now have *unpair*. The expression *unpair* (*x,y*)=*e1* *in* *e2* evaluates *e1*, makes sure it is a pair with components *v1* and *v2* and then evaluates *e2* in the extended environment where *x* is bound to *v1* and *y* to *v2*. Examples of programs in this extension are the first three examples in Exercise 2.2.3 and:

1. *unpair* (*x,y*) = *pair*(3, *pair*(5, 12)) *in* *x* is a program that evaluates to *Ok* (*NumVal* 3).
2. The program *let* *x* = 34 *in* *unpair* (*y,z*)=*pair*(2,*x*) *in* *z* evaluates to *Ok* (*NumVal* 34).

The abstract syntax of this extension is:

```

1 type expr =
2   ...
3   | Pair of expr*expr
4   | Unpair of string*string*expr*expr

```

ast.ml

Specify the interpreter (i.e. its evaluation rules) and then implement it.

**Exercise 2.2.5.** Consider the extension of LET with tuples. Its concrete syntax is that of LET together with the following new productions:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \text{tuple}(\langle \text{Expression} \rangle^{*(,)}) \\ \langle \text{Expression} \rangle &::= \text{untuple}(\langle \text{Identifier} \rangle^{*(,)}) = \langle \text{Expression} \rangle \text{ in } \langle \text{Expression} \rangle \end{aligned}$$

The  $*(,)$  above the nonterminal indicates zero or more copies separated by commas. The *tuple* keyword constructs a tuple with the values of its arguments. An expression of the form *untuple* (*x1*, ..., *xn*)=*e1* *in* *e2* first evaluates *e1*, makes sure it is a tuple of *n* values, say *v1* to *vn*, and then evaluates *e2* in the extended environment where each identifier *xi* is bound to *vi*. Examples of programs in this extension are:

1. *tuple* (2,3,4)
2. *tuple* (2,3,zero?(0))
3. *tuple* (*tuple*(7,9),3)
4. *tuple*(zero?(4),11-*x*)
5. *untuple* (*x,y,z*) = *tuple*(3, *tuple*(5, 12),4 ) *in* *x* is a program that evaluates to *Ok* (*NumVal* 3).
6. The program *let* *x* = 34 *in* *untuple* (*y,z*)=*tuple*(2,*x*) *in* *z* evaluates to *Ok* (*NumVal* 34).

Specify the interpreter (i.e. its evaluation rules) and then implement it.

**Exercise 2.2.6.** Consider the following extension of LET with records. Its concrete syntax is given adding the following new productions to that of LET:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \{ \langle \text{Identifier} \rangle = \langle \text{Expression} \rangle^{+(:)} \} \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle . \langle \text{Identifier} \rangle \end{aligned}$$

Examples of programs in this extension are:

1. `{age=2; height=3}`
2. `let person = {age=2; height=3} in let student = {pers=person; cwid=10} in student`
3. `{age=2; height=3}.age`
4. `{age=2; height=3}.ages`
5. `{age=2; age=3}`

Assume that the expressed values of LET are extended so that now records of expressed values may be produced as a result of evaluating a program (see the examples above).

```

2 type exp_val =
  ...
  | RecVal of (string*exp_val) list

```

The `expr` type encoding the AST is also extended:

```

2 type expr =
  ...
  | Record of (string*expr) list
4 | Proj of expr*string

```

Specify the interpreter (i.e. its evaluation rules) and then implement it. Some examples of the result of evaluation of this extension are:

1. `{age=2; height=3}`  
Should evaluate to `Ok (RecVal [("age", NumVal 2); ("height", NumVal 3)])`.
2. `let person = {age=2; height=3} in let student = {pers=person; cwid=10} in student`  
Should evaluate to `Ok (RecVal [("pers", RecVal [("age", NumVal 2); ("height", NumVal 3)]); ("cwid", NumVal 10)])`
3. `{age=2; height=3}.age`  
Should evaluate to `Ok (NumVal 2)`.
4. `{age=2; height=3}.ages`  
Should evaluate to `Error "Field not found"`.
5. `{age=2; age=3}`  
Should evaluate to `Error "Record has duplicate fields"`.

## 2.3 PROC

### 2.3.1 Concrete Syntax

$$\begin{aligned}
 \langle \text{Program} \rangle &::= \langle \text{Expression} \rangle \\
 \langle \text{Expression} \rangle &::= \langle \text{Number} \rangle \\
 \langle \text{Expression} \rangle &::= \langle \text{Identifier} \rangle \\
 \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle \langle \text{BOp} \rangle \langle \text{Expression} \rangle \\
 \langle \text{Expression} \rangle &::= \text{zero?}(\langle \text{Expression} \rangle) \\
 \langle \text{Expression} \rangle &::= \text{if } \langle \text{Expression} \rangle \text{ then } \langle \text{Expression} \rangle \text{ else } \langle \text{Expression} \rangle \\
 \langle \text{Expression} \rangle &::= \text{let } \langle \text{Identifier} \rangle = \langle \text{Expression} \rangle \text{ in } \langle \text{Expression} \rangle \\
 \langle \text{Expression} \rangle &::= (\langle \text{Expression} \rangle) \\
 \langle \text{Expression} \rangle &::= \text{proc}(\langle \text{Identifier} \rangle) \{ \langle \text{Expression} \rangle \} \\
 \langle \text{Expression} \rangle &::= (\langle \text{Expression} \rangle \langle \text{Expression} \rangle) \\
 \langle \text{BOp} \rangle &::= + \mid - \mid * \mid /
 \end{aligned}$$

### 2.3.2 Abstract Syntax

```

type expr =
2   | Var of string
   | Int of int
4   | Add of expr*expr
   | Sub of expr*expr
6   | Mul of expr*expr
   | Div of expr*expr
8   | Let of string*expr*expr
   | IsZero of expr
10  | ITE of expr*expr*expr
   | Proc of string*expr
12  | App of expr*expr

```

### 2.3.3 Interpreter

#### Specification

Evaluation judgements for PROC are exactly the same as for LET except that now the value resulting from evaluation of non-error computations, namely the expressed values, may either be an integer, a boolean, unit or a **closure**. A closure is a triple consisting of an identifier, an expression and an environment. All three sets, expressed values, closures and environments must be defined mutually recursively since they depend on each other:

$$\begin{aligned}
 \mathbb{EV} &:= \mathbb{Z} \cup \mathbb{B} \cup \mathbb{U} \cup \mathbb{CL} \\
 \mathbb{CL} &:= \{(\text{id}, e, \rho) \mid e \in \mathbb{EXP}, \text{id} \in \mathbb{ID}, \rho \in \mathbb{ENV}\} \\
 \mathbb{ENV} &:= \mathbb{ID} \rightarrow \mathbb{EV}
 \end{aligned}$$

The evaluation judgement for PROC reads:

$$e, \rho \Downarrow r$$

$$\begin{array}{c}
\frac{}{\text{Proc}(\text{id}, \text{e}), \rho \Downarrow (\text{id}, \text{e}, \rho)} \text{EProc} \\
\\
\frac{\text{e1}, \rho \Downarrow (\text{id}, \text{e}, \sigma) \quad \text{e2}, \rho \Downarrow w \quad \text{e}, \sigma \oplus \{\text{id} := w\} \Downarrow v}{\text{App}(\text{e1}, \text{e2}), \rho \Downarrow v} \text{EApp} \\
\\
\frac{\text{e1}, \rho \Downarrow v \quad v \notin \mathbb{CL}}{\text{App}(\text{e1}, \text{e2}), \rho \Downarrow \text{error}} \text{EAppErr}
\end{array}$$

**Figure 2.5:** Evaluation rules for PROC

The evaluation rules include those of LET (see Fig. 2.1) plus the additional rules given in Fig. 2.5.

### Implementation

To extend the interpreter for LET to PROC, we need to model closures and then extend `eval_expr`. Modeling closures as runtime values is easy since closures are simply triples consisting of an identifier, an expression and an environment:

```

type exp_val =
2 | NumVal of int
  | BoolVal of bool
4 | UnitVal
  | ProcVal of string*Ast.expr*env
6 and
  env =
8 | EmptyEnv
  | ExtendEnv of string*exp_val*env

```

ds.ml

Now, for `eval_expr`, we add code for two new variants in the definition of `eval_expr`, namely `Proc(id,e)` and `App(e1,e2)`. Let us first analyze the former. Our first attempt might look something like this:

```

let rec eval_expr : expr -> exp_val ea_result =
2 fun e ->
  match e with
4 | Proc(id,e) ->
    return (ProcVal(id,e, en))

```

Evaluation of `Proc(id,e)` should produce a closure that includes both of `id` and `e`. It must also include the current environment, denoted `en` above. However, the identifier `en` is not in scope. Indeed, the current environment is passed around in the background by the helper functions for `ea_result`. We introduce a new helper function that reads the current environment and returns it as a result (*i.e.* `Ok env`, where `env` is the environment).

```

let rec lookup_env : env ea_result =
2 fun env -> Ok env

```



ds.ml

With this new function we can now implement the evaluator for `Proc(id,e)`:

```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   | Proc(id,e) ->
        lookup_env >>= fun en ->
6       return (ProcVal(id,e,en))

```

interp.ml

Two alternative implementations for the `Proc(e1,e2)` case might be:

```

| Proc(id,e) ->
2   fun env -> return (ProcVal(id,e,env)) env

```

and

```

| Proc(id,e) ->
2   fun env -> Ok (ProcVal(id,e,env))

```

However these reveal the internal encoding of `exp_val ea_result`. The constructor `Ok` should not be used outside `ds.ml`. Likewise, the fact that `exp_val ea_result` is a function type should not be known outside `ds.ml`.

We next consider the case for `App(e1,e2)`. Evaluation of an application requires that we first evaluate `e1` obtaining an expressed value `v1`, then `e2` obtaining `v2`, and then we have to apply `v1` to `v2`. This last step implies checking that `v1` is indeed a closure and then evaluating the corresponding body. It is achieved by means of the helper function `apply_proc`:

```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   | App(e1,e2) ->
        eval_expr e1 >>= fun v1 ->
6       eval_expr e2 >>= fun v2 ->
        apply_proc v1 v2

```

The function `apply_proc` below matches its first argument `f` with the pattern `ProcVal(id,body,env)` to see if it is a closure, returning an error if this is not the case. If it is, however, we must extend the environment `env` stored inside the closure, with a new binding for the parameter, and then evaluate the body. The expression `return env` returns the environment `env` as a result. We feed that environment into `extend_env id a` and then evaluate the body under the extended environment.

```

let rec apply_proc : exp_val -> exp_val -> exp_val ea_result =
2   fun f a ->
    match f with
4   | ProcVal(id,body,env) ->
        return env >>+
6       extend_env id a >>+
        eval_expr body
8   | _ -> error "apply_proc: Not a procVal"

```

Fig. 2.6 summarizes the code described above for procedures and applications.

```

1 let rec apply_proc : exp_val -> exp_val -> exp_val ea_result =
2   fun f a ->
3     match f with
4     | ProcVal(id,body,env) ->
5       return env >>+
6       extend_env id a >>+
7       eval_expr body
8     | _ -> error "apply_proc: Not a procVal"
9 and
10 eval_expr : expr -> exp_val ea_result =
11   fun e ->
12     match e with
13     | Proc(id,e) ->
14       lookup_env >>= fun en ->
15       return (ProcVal(id,e,en))
16     | App(e1,e2) ->
17       eval_expr e1 >>= fun v1 ->
18       eval_expr e2 >>= fun v2 ->
19       apply_proc v1 v2

```

interp.ml

Figure 2.6: Interpreter for PROC

### 2.3.4 Dynamic Scoping

If we comment out the line below, then we implement dynamic scoping:

```

1 let rec apply_proc : exp_val -> exp_val -> exp_val ea_result =
2   fun f a ->
3     match f with
4     | ProcVal(id,body,env) ->
5       (* return env >>+ *)
6       extend_env id a >>+
7       eval_expr body
8     | _ -> error "apply_proc: Not a procVal"

```

Indeed, in this case the environment that is extended by `extend_env id a` is the current environment and not the one that was saved in the closure. Here are some examples of executing programs in this variant of PROC:

```

1 interp "
2 let f = proc (x) { if zero?(x) then 1 else x*(f (x-1)) }
3 in (f 6) ";;
4 - : Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 720)

6 utop # interp "
7 let f = proc (x) { x+a }
8 in let a=2
9 in (f 2)";;
10 - : Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 4)

12 utop # interp "
13 let f= let a=2 in proc(x) { x+a}
14 in (f 2) ";;
15 - : Ds.exp_val Ds.result = Ds.Error "a not found!"

```

utop

## 2.4 Exercises

**Exercise 2.4.1.** Write a derivation to show that `let f = proc (x) { x-11 } in (f 77)` is a valid program in PROC.

**Exercise 2.4.2.** Write down the parse tree for the expression `let pred = proc(x) { x-1 } in (pred 5)`.

**Exercise 2.4.3.** Write down the result of evaluating the expressions below. Depict the full details of the closure, including the environment. Use the tabular notation seen in class to depict the environment.

- `proc (x) { x-11 }`
- `proc (x) { let y=2 in x }`
- `let a=1 in proc (x) { x }`
- `let a=1 in let b=2 in proc (x) { x }`
- `let f=(let b=2 in proc (x) { x }) in f`
- `proc (x) { proc (y) { x-y } }`

**Exercise 2.4.4.** Depict the environment extant at the breakpoint (signalled with the `debug` expression):

```

let a=1
2 in let b=2
  in let c=proc (x) { x }
4 in debug((c b))

```

**Exercise 2.4.5.** Depict the environment extant at the breakpoint:

```

let a=1
2 in let b=2
  in let c = proc (x) { debug(proc (y) { x-y } )}
4 in (c b)

```

**Exercise 2.4.6.** Depict the environment extant at the breakpoint:

```

let x=2
2 in let y=proc (d) { x }
  in let z=proc(d) { x }
4 in debug(3)

```

**Exercise 2.4.7** ( $\diamond$ ). Use the “higher-order” trick of self-application to implement the mutually recursive definitions of `even` and `odd` in PROC:

$$\begin{aligned}
 \text{even}(0) &= \text{true} \\
 \text{even}(n) &= \text{odd}(n-1) \\
 \\ 
 \text{odd}(0) &= \text{false} \\
 \text{odd}(n) &= \text{even}(n-1)
 \end{aligned}$$

**Exercise 2.4.8** ( $\diamond$ ). Use the “higher-order” trick of self-application to implement a function `pbt` that given a value  $v$  and a height  $h$  builds a perfect binary tree constructed out of pairs and that has  $v$  in the leaves and has height  $h$ . For example `((pbt 2) 3)` should produce

```
PairVal
2  (PairVal
   (PairVal (PairVal (NumVal 2, NumVal 2),
4     PairVal (NumVal 2, NumVal 2)),
   PairVal (PairVal (NumVal 2, NumVal 2),
6     PairVal (NumVal 2, NumVal 2))),
   PairVal
8  (PairVal (PairVal (NumVal 2, NumVal 2),
   PairVal (NumVal 2, NumVal 2)),
10 PairVal (PairVal (NumVal 2, NumVal 2),
   PairVal (NumVal 2, NumVal 2))))
```

### Exercise 2.4.9.

## 2.5 REC

Our language unfortunately does not support recursion<sup>3</sup>. The following attempt at defining factorial and then applying it to compute factorial of 5 fails. The reason is that `f` is not visible in the body of the `proc`.

```
let f =
2   proc (x) {
   if zero?(x)
4   then 1
   else x*(f (x-1)) }
6 in (f 5)
```

In order to verify this, evaluate the following expression:

```
let f =
2   proc (x) {
   debug(if zero?(x)
4   then 1
   else x*(f (x-1))) }
6 in (f 5)
```

Note that the environment in the closure for `f` does not include a reference to `f` itself. The next language we shall look at, namely REC, includes a new programming abstraction that allows us to define recursive functions. In REC we will write:

```
letrec fact(x) =
2   if zero?(x)
   then 1
4   else x * (fact (x-1))
in (fact 5)
```

<sup>3</sup>See exercises ?? and ?? on the “higher-order” trick though.

### 2.5.1 Concrete Syntax

```

⟨Program⟩      ::=  ⟨Expression⟩

⟨Expression⟩   ::=  ⟨Number⟩
⟨Expression⟩   ::=  ⟨Identifier⟩
⟨Expression⟩   ::=  ⟨Expression⟩⟨BOP⟩⟨Expression⟩
⟨Expression⟩   ::=  zero?(⟨Expression⟩)
⟨Expression⟩   ::=  if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩   ::=  let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩   ::=  (⟨Expression⟩)
⟨Expression⟩   ::=  proc(⟨Identifier⟩){⟨Expression⟩}
⟨Expression⟩   ::=  (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩   ::=  letrec ⟨Identifier⟩(⟨Identifier⟩) = ⟨Expression⟩ in ⟨Expression⟩

⟨BOP⟩          ::=  + | - | * | /

```

### 2.5.2 Abstract Syntax

```

1 type expr =
  | Var of string
3  | Int of int
  | Add of expr*expr
5  | Sub of expr*expr
  | Mul of expr*expr
7  | Div of expr*expr
  | Let of string*expr*expr
9  | IsZero of expr
  | ITE of expr*expr*expr
11 | Proc of string*expr
  | App of expr*expr
13 | Letrec of string*string*expr*expr

```

ast.ml

For example, the result of parsing the expression:

```

1 letrec fact(x) =
  if zero?(x)
3  then 1
  else x * (fact (x-1))
5 in (fact 5)

```

is the AST:

```

1 AProg
  (Letrec ("fact", "x",
3    ITE (IsZero (Var "x"), Int 1,
      Mul (Var "x", App (Var "fact", Sub (Var "x", Int 1))))),
5    App (Var "fact", Int 5))

```

$$\begin{array}{c}
\frac{\mathbf{e2}, \rho \oplus \{\mathbf{id} := (\mathbf{par}, \mathbf{e1}, \rho)^r\} \Downarrow v}{\mathbf{Letrec}(\mathbf{id}, \mathbf{par}, \mathbf{e1}, \mathbf{e2}), \rho \Downarrow v} \text{ELetRec} \\
\\
\frac{\rho(\mathbf{id}) = (\mathbf{par}, \mathbf{e}, \sigma)^r}{\mathbf{Var}(\mathbf{id}), \rho \Downarrow (\mathbf{par}, \mathbf{e}, \sigma \oplus \{\mathbf{id} := (\mathbf{par}, \mathbf{e}, \sigma)^r\})} \text{EVarLetRec}
\end{array}$$

---

**Figure 2.7:** Evaluation rules for REC

### 2.5.3 Interpreter

Recursive functions will be represented as special closures called recursion closures. Later we will look at another implementation involving circular environments. A **recursion closure** is a closure with a tag “*r*” to distinguish it from a standard closure, written  $(\mathbf{id}, \mathbf{e}, \rho)^r$ , where  $\mathbf{e} \in \text{EXP}$ ,  $\mathbf{id} \in \text{ID}$  and  $\rho \in \text{ENV}$ . The set of all recursion closures is denoted  $\text{RCL}$ :

$$\begin{array}{ll}
\text{ENV} & := \text{ID} \rightarrow \text{EV} \cup \text{RCL} \\
\text{EV} & := \text{Z} \cup \text{B} \cup \text{U} \cup \text{C} \cup \text{L} \\
\text{CL} & := \{(\mathbf{id}, \mathbf{e}, \rho) \mid \mathbf{e} \in \text{EXP}, \mathbf{id} \in \text{ID}, \rho \in \text{ENV}\} \\
\text{RCL} & := \{(\mathbf{id}, \mathbf{e}, \rho)^r \mid \mathbf{e} \in \text{EXP}, \mathbf{id} \in \text{ID}, \rho \in \text{ENV}\}
\end{array}$$

Note that recursion closures are not expressed values. We cannot write a program that, when evaluated, returns a recursion closure. They are an auxiliary device for defining evaluation of recursive programs. More precisely, recursive function definitions will be stored as recursion closures. However, lookup of recursive functions will produce standard closures, the latter being computed on the fly.

#### Specification

The set of results are:

$$\mathbb{R} := \text{EV} \cup \{\text{error}\}$$

Evaluation judgements for REC are the same as for PROC:

$$\mathbf{e}, \rho \Downarrow r$$

where  $r \in \mathbb{R}$ . Evaluation rules for REC are those of PROC together with the ones in Fig. 2.7. Two new evaluation rules are added to those of PROC to obtain REC. The rule  $\text{ELetRec}$  creates a recursion closure and adds it to the current environment  $\rho$  and then continues with the evaluation of  $\mathbf{e2}$ . The rule  $\text{EVarLetRec}$  does lookup of identifiers that refer to previously declared recursive functions. Upon finding the corresponding recursion closure in the current environment, it creates a new closure and returns it. Note that the newly created closure includes an environment that has a reference to  $\rho$  itself.

#### Implementation

Recursion closures are implemented by adding a new constructor to the type `expr`, namely `ExtendEnvRec` below:

```

type exp_val =
2 | NumVal of int
  | BoolVal of bool
4 | ProcVal of string*Ast.expr*env
  | UnitVal
6 and
  env =
8 | EmptyEnv
  | ExtendEnv of string*exp_val*env
10 | ExtendEnvRec of string*string*Ast.expr*env

```

ds.ml

Note that the arguments of `ExtendEnvRec(id,par,body,env)` are four: the name of the recursive function being defined `id`, the name of the formal parameter `par`, the body of the recursive function `body`, and the rest of the environment `env`. If we consider the environment  $\rho \oplus \{id := (par, e1, \rho)\}$  in the rule `ELetRec` of Fig. 2.7, it would seem we are missing an argument. Indeed, the operator “ $\oplus$ ” in the evaluation rule is modeled by the `ExtendEnvRec` constructor in our implementation. However, there is no need to store  $\rho$  in our implementation since it is just the tail of the environment.

Next we need an operation similar to `extend_env` but that adds a new recursion closure to the environment:

```

let extend_env_rec : string -> string -> Ast.expr -> env ea_result =
2 fun id par body ->
  fun env -> Ok (ExtendEnvRec(id,par,body,env))

```

ds.ml

In addition, we need to update the implementation of `apply_env` so that it deals with lookup of recursive functions, thus correctly implementing `EVarLetRec`. This involves adding a new clause (see code highlighted below):

```

let rec apply_env : string -> exp_val ea_result =
2 fun id ->
  fun env ->
4 match env with
  | EmptyEnv -> Error (id^" not found!")
  | ExtendEnv(v,ev,tail) ->
6     if id=v
8     then Ok ev
     else apply_env id tail
10 | ExtendEnvRec(v,par,body,tail) ->
    if id=v
12 then Ok (ProcVal (par,body,env))
    else apply_env id tail

```

ds.ml

Regarding the code for the interpreter itself, we need only add a new clause, namely the one for `Letrec(id,par,e1,e2)`:

```

| Letrec(id,par,e1,e2) ->
2   extend_env_rec id par e1 >>+
   eval_expr e2

```

interp.ml

**Exercise 2.5.1.** Evaluate the following expressions in *utop*:

```
1.
1  utop # interp "
   let one=1
3  in letrec fact(x) =
      if zero?(x)
5      then one
      else x * (fact (x-1))
7  in debug((fact 6))" ;;
```

```
2.
1  utop # interp "
   let one=1
3  in letrec fact(x) =
      debug(if zero?(x)
5      then one
      else x * (fact (x-1)))
7  in (fact 6)" ;;
```

```
3.
1  utop # interp "
   let one=1
3  in letrec fact(x) =
      if zero?(x)
5      then one
      else x * (fact (x-1))
7  in fact" ;;
```

**Exercise 2.5.2** ( $\diamond$ ). Consider the following functions in *OCaml*:

```
1  let rec add n m =
   match n with
3  | 0 -> m
   | n' -> 1 + add (n'-1) m
5
7  let rec append l1 l2 =
   match l1 with
9  | [] -> l2
   | h::t -> h :: append t l2
11
13 let rec map l f =
   match l with
15 | [] -> []
   | h::t -> (f h) :: map t f
17
19 let rec filter l p =
   match l with
21 | [] -> []
   | h::t ->
       if p h
       then h :: filter t p
       else filter t p
23
25 let rec foldr l f a =
   match l with
27 | [] -> a
   | h::t -> f h (foldr t f a)
```



Code them in the extension of REC of Exercise 2.4.9. For example, here is the code for `add`:

```
# interp "
2 letrec add(n) = proc (m) {
      if zero?(n)
4       then m
      else 1 + ((add (n-1)) m) }
6 in ((add 2) 3)";;
- : exp_val Rec.Ds.result = Ok (NumVal 5)
```

utop

**Exercise 2.5.3** ( $\diamond$ ). Consider the following expression in REC

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

A `debug` instruction was placed somewhere in the code and it produced the environments below. Where was it placed? Identify and signal (see instructions below) the location for each of the three items below. Note that there may be more than one solution for each item, it suffices to supply just one.

1.

```
[
  z:=NumVal 0
  prod:=ProcVal(x, Proc(y, Mul(Var x, Var y)), [z:=NumVal 0])
  f:=Rec(n, IfThenElse(Zero?(Var n),
                      Int 1,
                      App(App(Var prod, Var n), App(Var f, Sub(Var n, Int 1))))))
  n:=NumVal 0
]
```

Draw a box around the argument of `debug`:

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

2.

```
[
  z:=NumVal 0
]
```

Draw a box around the argument of `debug`:

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

3.

```
[
  z:=NumVal 0
  x:=NumVal 10
]
```

Draw a box around the argument of *debug*:

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

## Chapter 3

# Imperative Programming

### 3.1 Mutable Data Structures in OCaml

#### 3.1.1 A counter object

```
type counter =  
2 { inc: unit -> unit;  
   dec: unit -> unit;  
4   read : unit -> int }
```

```
let c =  
2   let state = ref 0 in  
   { inc = (fun () -> state := !state+1);  
4     dec = (fun () -> state := !state-1);  
     read = (fun () -> !state) }
```

```
1 utop # c.read ();;  
- : int = 0  
3 utop # c.inc ();;  
- : unit = ()  
5 utop # c.read ();;  
- : int = 1  
7 utop # c.dec ();;  
- : unit = ()  
9 utop # c.read ();;  
- : int = 0
```

utop

#### 3.1.2 A stack object

```
type stack =  
2 { push : int -> unit;  
   pop : unit -> int;  
4   top : unit -> int };
```

```

let s = let state = ref []
2       in { push = (fun i -> state := i :: !state);
           pop = (fun () -> let temp = List.hd !state
4                           in state := List.tl !state; temp);
           top = (fun () -> List.hd !state)}

1 utop # s.push 1;;
- : unit = ()
3 utop # s.push 2;;
- : unit = ()
5 utop # s.pop ();;
- : int = 2
7 utop # s.top ();;
- : int = 1

```

utop

## 3.2 EXPLICIT-REFS

The following is an extension of REC.

### 3.2.1 Concrete Syntax

Examples of expressions in EXPLICIT-REFS:

```

newref(2)

2
let a=newref(2)
4 in a

6 let a=newref(2)
  in deref(a)

8
let a=newref(2)
10 in setref(a,deref(a)+1)

12 let a=newref(2)
   in begin
14     setref(a,deref(a)+1);
     deref(a)
16   end

18 let g =
   let counter = newref(0)
20   in proc (d) {
     begin
22       setref(counter, deref(counter)+1);
       deref(counter)
24     end
   }
26 in (g 11) - (g 22)

```

```

⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOp⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)
⟨Expression⟩ ::= proc(⟨Identifier⟩){⟨Expression⟩}
⟨Expression⟩ ::= (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩ ::= letrec⟨Identifier⟩(⟨Identifier⟩)=⟨Expression⟩in⟨Expression⟩
⟨Expression⟩ ::= newref(⟨Expression⟩)
⟨Expression⟩ ::= deref(⟨Expression⟩)
⟨Expression⟩ ::= setref(⟨Expression⟩,⟨Expression⟩)
⟨Expression⟩ ::= begin ⟨Expression⟩*(:) end

⟨BOp⟩          ::= + | - | * | /

```

The notation  $*(:)$  above the nonterminal  $\langle \text{Expression} \rangle$  in the production for begin/end indicates zero or more expressions separated by semi-colons.

### 3.2.2 Abstract Syntax

```

type expr =
2   | Var of string
   | Int of int
4   | Add of expr*expr
   | Sub of expr*expr
6   | Mul of expr*expr
   | Div of expr*expr
8   | Let of string*expr*expr
   | IsZero of expr
10  | ITE of expr*expr*expr
   | Proc of string*expr
12  | App of expr*expr
   | Letrec of string*string*expr*expr
14  | NewRef of expr
   | DeRef of expr
16  | SetRef of expr*expr
   | BeginEnd of expr list
18  | Debug of expr

```

### 3.2.3 Interpreter

#### Specification

We assume given a set of (symbolic) memory locations  $\mathbb{L}$ . We write  $\ell, \ell_i$  for memory locations. A heap or **store** is a partial function from memory locations to expressed values. The set of stores is denoted  $\mathbb{S}$ :

$$\mathbb{S} := \mathbb{L} \rightarrow \mathbb{EV}$$

$$\begin{array}{c}
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad \ell \notin \text{dom}(\sigma)}{\text{NewRef}(\mathbf{e}), \rho, \sigma \Downarrow \ell, \sigma' \oplus \{\ell := v\}} \text{ENewRef} \\
\\
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \in \text{dom}(\sigma')}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \sigma'(v), \sigma'} \text{EDeRef} \\
\\
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \notin \mathbb{L}}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{EDeRefErr1} \quad \frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \notin \text{dom}(\sigma')}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{EDeRefErr2} \\
\\
\frac{\mathbf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad \mathbf{e2}, \rho, \sigma' \Downarrow w, \sigma''}{\text{SetRef}(\mathbf{e1}, \mathbf{e2}), \rho, \sigma \Downarrow \text{unit}, \sigma'' \oplus \{v := w\}} \text{ESetRef} \quad \frac{\mathbf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \notin \mathbb{L}}{\text{SetRef}(\mathbf{e1}, \mathbf{e2}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{ESetRefErr} \\
\\
\frac{n > 0 \quad (\mathbf{e1}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1..n}}{\text{BeginEnd}([\mathbf{e1}; \dots; \mathbf{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \text{EBeginEndNE} \\
\\
\frac{}{\text{BeginEnd}([]), \rho, \sigma \Downarrow \text{unit}, \sigma} \text{EBeginEndE}
\end{array}$$

---

**Figure 3.1:** Evaluation rules for EXPLICIT-REFS

where the set of expressed values includes locations:

$$\text{EV} := \mathbb{Z} \cup \mathbb{B} \cup \mathbb{U} \cup \mathbb{C} \cup \mathbb{L} \cup \mathbb{L}$$

Evaluation judgements in EXPLICIT-REFS take the following form, where  $\mathbf{e}$  is an expression,  $\rho$  and environment,  $\sigma$  the initial store,  $r$  the result and  $\sigma'$  the final store

$$\mathbf{e}, \rho, \sigma \Downarrow r, \sigma'$$

Note that the result of evaluating an expression now returns both a result and an updated store. The evaluation rules for EXPLICIT-REFS are given in Fig. 3.1.

### Implementing Stores

The implementation of the evaluator for EXPLICIT-REFS requires that we first implement stores. Since a store is a mutable data structure we will use OCaml arrays. The following interface file declares the types of the values in the public interface of the store. These values include a parametric type constructor `Store.t`, the type of the store itself and multiple functions.

```

1 type 'a t
2
3 val empty_store : int -> 'a -> 'a t
4 val get_size : 'a t -> int
5 val new_ref : 'a t -> 'a -> int
6 val deref : 'a t -> int -> 'a option
7 val set_ref : 'a t -> int -> 'a -> unit option

```

```
8 val string_of_store : ('a -> string) -> 'a t -> string
```

store.mli

These operations are:

- `empty_store n v` returns a store of size `n` where each element is initialized to `v`
- `get_size s` returns the number of elements in the store.
- `new_ref s v` stores `v` in a fresh location and returns the location.
- `deref s l` returns the contents of location `l`, prefixed by `Some`, in the store `s`. This operation fails, returning `None`, if the location is out of bounds.
- `set_ref s l v` updates the contents of `l` in `s` with `v`. It fails, returning `None`, if the index is out of bounds.
- `string_of_store to_str s` returns a string representation of `s` resulting from applying `to_str` to each element.

Each of the above operations implemented in `store.ml`.

```
open Ds
2
type 'a t = { mutable data: 'a array; mutable size: int }
4 (* data is declared mutable so the store may be resized *)

6 let empty_store : int -> 'a -> 'a t =
  fun i v -> { data=Array.make i v; size=0 }

8
let get_size : 'a t -> int =
10 fun st -> st.size

12 let enlarge_store : 'a t -> 'a -> unit =
  fun st v ->
14   let new_array = Array.make (st.size*2) v
  in Array.blit st.data 0 new_array 0 st.size;
16   st.data<-new_array

18 let new_ref : 'a t -> 'a -> int =
  fun st v ->
20   if Array.length (st.data)=st.size
  then enlarge_store st v
22   else ();
  begin
24     st.data.(st.size)<-v;
     st.size<-st.size+1;
26     st.size-1
  end

28

30 let deref : 'a t -> int -> 'a option =
  fun st l ->
32   if l>=st.size
  then None (* Index out of bounds *)
34   else Some (st.data.(l))
```

```

36 let set_ref : 'a t -> int -> 'a -> unit option =
    fun st l v ->
38   if l >= st.size
    then None (* Index out of bounds *)
40   else Some (st.data.(l) <- v)

42
44 let rec take n = function
    | [] -> []
    | x::xs when n > 0 -> x::take (n-1) xs
46   | _ -> []

48 let string_of_store' f st =
    let ss = List.mapi (fun i x -> (i, f x)) @@ take st.size @@ Array.to_list st.data
50   in
    if ss == []
52   then "Empty"
    else List.fold_left (fun curr (i,s) -> curr ^ string_of_int i ^ "->" ^ s ^ "\n")
54     ""
    ss
56
58 let string_of_store f st =
    "Store:\n" ^ string_of_store' f st

```

store.ml

### Implementation

We could have followed the ideas we developed for environments and have them be threaded for us behind the scenes. This would lead to a similar extension of our current result type `ea_result` so that it also abstracts over the store. Consequently, we would avoid having to thread the store around. However, in order to keep things simple and since the concept of threading behind the scenes has already been introduced via environments, we choose to hold the store in a top-level or global variable `g_store`.

```
let g_store = Store.empty_store 20 (NumVal 0)
```

interp.ml

`g_store` denotes a store of size 20, whose values have arbitrarily been initialized to `NumVal 0`.

Next we consider the new expressed value, namely symbolic locations. Such locations will be denoted by an integer wrapped inside a `RefVal` constructor:

```

2 type exp_val =
    | NumVal of int
    | BoolVal of bool
    | ProcVal of string * Ast.expr * env
    | UnitVal
6   | RefVal of int

```

ds.ml

Next we move on to the interpreter, only addressing the new variants.

```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   | NewRef(e) ->

```



```

6   eval_expr e >>= fun ev ->
    return @@ RefVal (Store.new_ref g_store ev)
| DeRef(e) ->
    eval_expr e >>=
    int_of_refVal >>= fun l ->
10  (match Store.deref g_store l with
    | None -> error "index out of bounds"
    | Some ev -> return ev)
| SetRef(e1,e2) ->
14  eval_expr e1 >>=
    int_of_refVal >>= fun l ->
16  eval_expr e2 >>= fun ev ->
    (match Store.set_ref g_store l ev with
18  | None -> error "index out of bounds"
    | Some _ -> return UnitVal)
| BeginEnd([]) ->
    return UnitVal
22 | BeginEnd(es) ->
    sequence (List.map eval_expr es) >>= fun vs ->
24  return (List.hd (List.rev vs))
| Debug(_e) ->
26  string_of_env >>= fun str_env ->
    let str_store = Store.string_of_store string_of_expval g_store
28  in (print_endline (str_env^str_store);
    return UnitVal)
30 | _ -> error ("Not implemented: "^string_of_expr e)

```

[interp.ml](#)

## 3.3 IMPLICIT-REFS

The following is an extension of REC.

### 3.3.1 Concrete Syntax

Examples of expressions in IMPLICIT-REFS

```

1  let x=2
   in begin
3    set x=3;
    x
5  end

7  let x=2
   let y=x+1
   in begin
9    set y=y+1;
    y
11 end

13 let x=2
15 in let f = proc (n) { begin set x=x+1; 1 end }
   in let g = proc (n) { begin set x=x+1; 2 end }
17 in begin
    (f 0)+(g 0);
19  x

```

end

```

<Expression> ::= <Number>
<Expression> ::= <Identifier>
<Expression> ::= <Expression> <BOp> <Expression>
<Expression> ::= zero?(<Expression>)
<Expression> ::= if <Expression> then <Expression> else <Expression>
<Expression> ::= let <Identifier> = <Expression> in <Expression>
<Expression> ::= (<Expression>)
<Expression> ::= proc(<Identifier>){<Expression>}
<Expression> ::= (<Expression> <Expression>)
<Expression> ::= letrec <Identifier>(<Identifier>)=<Expression> in <Expression>
<Expression> ::= set(<Expression>, <Expression>)
<Expression> ::= begin <Expression>+(;) end

<BOp> ::= + | - | * | /

```

### 3.3.2 Abstract Syntax

```

2 type expr =
  | Var of string
  | Int of int
  | Add of expr*expr
  | Sub of expr*expr
  | Mul of expr*expr
  | Div of expr*expr
  | Let of string*expr*expr
  | IsZero of expr
  | ITE of expr*expr*expr
  | Proc of string*expr
  | App of expr*expr
  | Letrec of string*string*expr*expr
  | Set of string*expr
  | BeginEnd of expr list
16 | Debug of expr

```

### 3.3.3 Interpreter

#### Specification

Since in IMPLICIT-REFS all identifiers are mutable, the environment will map all identifiers to locations in the store. Evaluation judgements in IMPLICIT-REFS take the following form, where  $e$  is an expression,  $\rho$  and environment,  $\sigma$  the initial store,  $r$  the result and  $\sigma'$  the final store

$$e, \rho, \sigma \Downarrow r, \sigma'$$

As mentioned,  $\rho$  maps identifiers to locations, it **no longer** maps them to expressed values. Hence identifier lookup now has to lookup the location first in the environment and then access the contents in the store. This is exactly what the rule EVar states:

$$\begin{array}{c}
\frac{\sigma(\rho(\text{id})) = v}{\text{Var}(\text{id}), \rho, \sigma \Downarrow v, \sigma} \text{EVar} \quad \frac{\rho(\text{id}) \notin \mathbb{L} \text{ or } \rho(\text{id}) \notin \text{dom}(\sigma)}{\text{Var}(\text{id}), \rho, \sigma \Downarrow \text{error}, \sigma} \text{EVarErr} \\
\\
\frac{}{\text{Proc}(\text{id}, \text{e}), \rho, \sigma \Downarrow (\text{id}, \text{e}, \rho), \sigma} \text{EProc} \\
\\
\frac{\text{e1}, \rho, \sigma \Downarrow (\text{id}, \text{e}, \tau), \sigma_1 \quad \text{e2}, \rho, \sigma_1 \Downarrow w, \sigma_2 \quad \ell \notin \text{dom}(\sigma_1) \quad \text{e}, \tau \oplus \{\text{id} := \ell\}, \sigma_2 \oplus \{\ell := w\} \Downarrow v, \sigma_3}{\text{App}(\text{e1}, \text{e2}), \rho, \sigma \Downarrow v, \sigma_3} \text{EApp} \\
\\
\frac{\text{e}, \rho, \sigma \Downarrow v, \sigma'}{\text{Set}(\text{id}, \text{e}), \rho, \sigma \Downarrow \text{unit}, \sigma' \oplus \{\rho(\text{id}) := v\}} \text{ESet} \quad \frac{\rho(\text{id}) \notin \mathbb{L} \text{ or } \rho(\text{id}) \notin \text{dom}(\sigma)}{\text{Set}(\text{id}, \text{e}), \rho, \sigma \Downarrow \text{error}, \sigma} \text{ESetErr} \\
\\
\frac{n > 0 \quad (\text{ei}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1..n}}{\text{BeginEnd}([\text{e1}; \dots; \text{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \text{EBeginEndNE} \\
\\
\frac{}{\text{BeginEnd}([]), \rho, \sigma \Downarrow \text{unit}, \sigma} \text{EBeginEndE}
\end{array}$$

---

**Figure 3.2:** Evaluation rules for IMPLICIT-REFS

$$\frac{\sigma(\rho(\text{id})) = v}{\text{Var}(\text{id}), \rho, \sigma \Downarrow v, \sigma} \text{EVar}$$

Indeed,  $\rho(\text{id})$  denotes a location whose contents is looked up in the store  $\sigma$ . If  $\rho(\text{id})$  is not a valid location, then an error is returned, as described by rule EVarErr. The full set of evaluation rules are given in Fig. 3.2.

### Implementation

We address the implementation of the evaluator. For now we ignore `Letrec` and then take it up later. Instead we focus on the `App(e1, e2)` case, which needs some minor updating, and also on the new variants.

Regarding the `App(e1, e2)` case, we need to slightly modify the `apply_proc` function. We briefly recall the code for `apply_proc` as implemented in the PROC (Fig. 2.6):

```

let rec apply_proc : exp_val -> exp_val -> exp_val ea_result =
2   fun f a ->
   match f with
4   | ProcVal(id, body, env) ->
       return env >>+
       extend_env id a >>+
       eval_expr body
8   | _ -> error "apply_proc: Not a procVal"

```

Note that evaluation of the body requires extending the environment with a new key-value pair, namely  $(\text{id}, a)$ , where  $a$  is the expressed value supplied as argument. Environments no longer map

identifiers to expressed values, but to locations. So we first need to allocate `a` in the store in a fresh location `l` and then extend the environment with the key-value pair  $(id, l)$ . The updated code for `apply_proc` is given below.

```

let rec apply_proc ev1 ev2 =
2   match ev1 with
    | ProcVal(id,body,en) ->
4       return en >>+
        extend_env id (RefVal (Store.new_ref g_store ev2)) >>+
6       eval_expr body
    | _ -> error "apply_proc: Not a procVal"

```

We now address the new cases (and the ones we need to modify) for the interpreter:

```

1   | Var(id) ->
    apply_env id >>=
3   int_of_refVal >>= fun l ->
    (match Store.deref g_store l with
5   | None -> error "Index out of bounds"
    | Some ev -> return ev)
7   ...
    | Let(v,def,body) ->
9       eval_expr def >>= fun ev ->
        let l = Store.new_ref g_store ev
11      in extend_env v (RefVal l) >>+
        eval_expr body
13    | Set(id,e) ->
        eval_expr e >>= fun ev ->
15        apply_env id >>=
            int_of_refVal >>= fun l ->
17        (match Store.set_ref g_store l ev with
            | None -> error "Index out of bounds"
            | Some _ -> return UnitVal)
19    | BeginEnd([]) ->
        return UnitVal
21    | BeginEnd(es) ->
        sequence (List.map eval_expr es) >>= fun vs ->
23        return (List.hd (List.rev vs))
25    | Debug(_e) ->
        string_of_env >>= fun str_env ->
27        let str_store = Store.string_of_store string_of_expval g_store
        in (print_endline (str_env^str_store));
        return UnitVal
29    | _ -> error ("Not implemented: "^string_of_expr e)

```

### letrec Revisited

Our implementation of `letrec` in REC consisted in adding a specific entry in the environment to signal the declaration of a recursive function. Then, upon lookup, a closure was created on the fly. This is the code from REC. The highlighted excerpt `Ok (ProcVal (par,body,env))` indicates that a closure is being created.

```

type env =
2   | EmptyEnv
    | ExtendEnv of string*exp_val*env
4   | ExtendEnvRec of string*string*Ast.expr*env

```

```

6 let rec apply_env : string -> exp_val ea_result =
  fun id env ->
8   match env with
  | EmptyEnv -> Error (id ^ " not found!")
10  | ExtendEnv(v, ev, tail) ->
    if id=v
12    then Ok ev
    else apply_env id tail
14  | ExtendEnvRec(v, par, body, tail) ->
    if id=v
16    then Ok (ProcVal (par, body, env))
    else apply_env id tail

```

ds.ml

We could follow the same approach in IMPLICIT-REFS, but that would require `apply_env` to access the store so that it could allocate space for the closure created on the fly. This is not very neat: why would looking up a value in the environment involve using the store? An alternative approach is simply to allow circular environments. That is, an environment `env` that has an entry which is a reference into the store that contains a closure whose environment is `env` itself.

So we first remove the special entry in environments for `letrec` declarations since they will no longer be needed:

```

type env =
2   | EmptyEnv
  | ExtendEnv of string*exp_val*env
4   | ExtendEnvRec of string*string*Ast.expr*env

6 let rec apply_env : string -> exp_val ea_result =
  fun id env ->
8   match env with
  | EmptyEnv -> Error (id ^ " not found!")
10  | ExtendEnv(v, ev, tail) ->
    if id=v
12    then Ok ev
    else apply_env id tail
14  | ExtendEnvRec(v, par, body, tail) ->
    if id=v
16    then Ok (ProcVal (par, body, env))
    else apply_env id tail

```

ds.ml

We now use “back-patching” to code the circular environment:

```

1 let rec eval_expr : expr -> exp_val ea_result =
  fun e ->
3   match e with
  | Letrec(id, par, e, target) ->
5     let l = Store.new_ref g_store UnitVal in
      extend_env id (RefVal l) >>+
7     (lookup_env >>= fun env ->
        (let[@warning "-8"] Some _ = Store.set_ref g_store l (ProcVal(par, e, env))
          in eval_expr target)
      )
9

```

interp.ml



Parenthesis right after `>>+` are necessary since `>>=` and `>>+` are left-associative. Remove them, execute the resulting interpreter on an example expression and explain what goes wrong.

## 3.4 Parameter Passing Methods

### 3.4.1 Call-by-Value

### 3.4.2 Call-by-Reference

### 3.4.3 Call-by-Name

### 3.4.4 Call-by-Need

## 3.5 Exercises

**Exercise 3.5.1.** Consider the following extension of LET with records (Exercise 2.2.6). It has the same syntax except that one can declare a field to be mutable by using `<=` instead of `=`. For example, the `ssn` field is immutable but the `age` field is mutable; `age` is then updated to 31:

```

1 let p = {ssn = 10; age <= 30}
2 in begin
3   p.age <= 31;
4   p.age
end

```

Evaluating this expression should produce `Ok (NumVal 31)`. This other expression should produce `Ok (RecordVal [("ssn", (false, NumVal 10)); ("age", (true, RefVal 1))])`:

```

1 let p = {ssn = 10; age <= 30}
2 in begin
3   p.age <= 31;
4   p
end

```

Updating an immutable field should not be allowed. For example, the following expression should report an error `Error "Field not mutable"`:

```

1 let p = { ssn = 10; age = 20}
2 in begin
3   p.age <= 21;
4   p.age
end

```

The abstract syntax requires modifying the `Record` constructor and adding a new one for field update:

```

1 type expr =
2   ...
3   | Record of (string*(bool*expr)) list
4   | SetField of expr*string*expr

```

For example,

```

# parse "
2 let p = {ssn = 10; age <= 30}
in begin
4   p.age <= 31;
   p
6   end";;
- : expr =
8 Let ("p", Record [("ssn", (false, Int 10)); ("age", (true, Int 30))],
   BeginEnd [SetField (Var "p", "age", Int 31); Var "p"])

```

utop

Here *false* indicates the field is immutable and *true* that it is mutable. You are asked to implement the interpreter extension. The *RecordVal* constructor has been updated for you.

```

type exp_val =
2   | NumVal of int
   | BoolVal of bool
4   | ProcVal of string*Ast.expr*env
   | PairVal of exp_val*exp_val
6   | TupleVal of exp_val list
   | RecordVal of (string*(bool*exp_val)) list

```

ds.ml

As for *eval\_expr*, the case for *Record* has already been updated for you. You are asked to update *Proj* and complete *SetField*:

```

let rec eval_expr : expr -> exp_val ea_result = fun e ->
2   match e with
   | Record(fs) ->
4     sequence (List.map process_field fs) >>= fun evs ->
       return (RecordVal (addIds fs evs))
   | Proj(e,id) ->
6     error "update"
   | SetField(e1,id,e2) ->
8     error "implement"
10  and
   process_field (_id,(is_mutable,e)) =
12    eval_expr e >>= fun ev ->
       if is_mutable
14    then return (RefVal (Store.new_ref g_store ev))
       else return ev

```

**Exercise 3.5.2.** Depict the environment and store that is extant at the breakpoint.

```

1 let a = 2
in let b = 3
3 in begin
   set a = b;
5   debug(a)
   end

```

**Exercise 3.5.3.** Depict the environment and store that is extant at the breakpoint.

```

1 let a = 2
2 in let b = a

```

```
in begin
4   set b = 3;
   debug(a)
6   end
```

**Exercise 3.5.4.** *Depict the environment and store that is extant at the breakpoint.*

```
let a = 2
2 in let b = proc(x) {
   begin
4     set a = x;
     debug(a)
6     end
   }
8 in (b 3)
```

**Exercise 3.5.5.** *Depict the environment and store that is extant at the breakpoint.*

```
let a = 2
2 in let b = proc(x) {
   begin
4     set a = x;
     a
6     end
   }
8 in (b 3) + debug((b 4))
```



# Chapter 4

## Types

This chapter extends the REC language to support type-checking.

### 4.1 CHECKED

#### 4.1.1 Concrete Syntax

```
⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOP⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)
⟨Expression⟩ ::= proc(⟨Identifier⟩ : ⟨Type⟩){⟨Expression⟩}
⟨Expression⟩ ::= (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩ ::= letrec⟨Identifier⟩(⟨Identifier⟩ : ⟨Type⟩) : ⟨Type⟩=⟨Expression⟩in⟨Expression⟩

⟨BOP⟩ ::= + | - | * | /

⟨Type⟩ ::= int
⟨Type⟩ ::= bool
⟨Type⟩ ::= ⟨Type⟩->⟨Type⟩
⟨Type⟩ ::= (⟨Type⟩)
```

#### 4.1.2 Abstract Syntax

```
type expr =
2 | Var of string
  | Int of int
4 | Sub of expr*expr
  | Let of string*expr*expr
```

```

6 | IsZero of expr
  | ITE of expr*expr*expr
8 | Proc of string*texpr*expr
  | App of expr*expr
10 | Letrec of string*string*texpr*texpr*expr*expr
and
12 texpr =
  | IntType
  | BoolType
14 | FuncType of texpr*texpr

```

### 4.1.3 Type-Checker

We specify the behavior of our type-checker before implementing it, much like we do with interpreters. For this task we use **type systems**. A type system is an inductive set that helps identify a subset of the expressions that are considered to be well-typed. The elements of the inductive set are called **type judgements**. Which type judgements belong to the set and which don't is determined by a set of type rules. A type judgement is an expression of the form

$$\Gamma \vdash e : t$$

where  $\Gamma$  is a type environment,  $e$  is an expression in CHECKED, and  $t$  is a type expression. These components together with the type rules are introduced below.

#### Specification

As mentioned, a type judgement consists of a of type environment, an expression in CHECKED and a type. Types are defined as follows:

$$t ::= \text{int} \mid \text{bool} \mid t \rightarrow t$$

A **type context** is a partial function that assigns a type to an identifier. Type contexts are required for typing expressions that contain free variables. For example, an expression such as  $x+2$  will require that we have the type of  $x$  at our disposal in order to determine whether  $x+2$  is typable at all. If the type of  $x$  were `bool`, then it is not typable; but if the type of  $x$  is `int`, then it is. Type contexts are defined as follows:

$$\Gamma ::= \epsilon \mid \Gamma, id : t$$

We use  $\epsilon$  to denote the empty type environment. Also,  $\Gamma, id : t$  assigns type  $t$  to identifier  $id$  and behaves as  $\Gamma$  for identifiers different from  $id$ . We assume that  $\Gamma$  does not have repeated entries for the same identifier. An example of a type contexts is  $\epsilon, x : \text{int}, y : \text{bool}$ . We abbreviate it as  $x : \text{int}, y : \text{bool}$ .

The type rules are given in Fig. 4.1.

#### Towards and Implementation

Our type checker will behave very much like our interpreter, except that instead of manipulating runtime values such as integers and booleans, it manipulates types like `int` and `bool`. One might say that a type checker is a symbolic evaluator, where our symbolic values are the types. This analogy allows us to apply the ideas we have developed on well-structuring an evaluator to our type checker. Thus one might be tempted to state the type of our type-checker as

$$\begin{array}{c}
\frac{}{\Gamma \vdash n : \text{int}} \text{TConst} \quad \frac{\Gamma(x) = t}{\Gamma \vdash x : t} \text{TVar} \quad \frac{\Gamma \vdash e : \text{int}}{\Gamma \vdash \text{zero?}(e) : \text{bool}} \text{TZero} \\
\\
\frac{\Gamma \vdash e1 : \text{int} \quad \Gamma \vdash e2 : \text{int}}{\Gamma \vdash e1 - e2 : \text{int}} \text{TDiff} \\
\\
\frac{\Gamma \vdash e1 : \text{bool} \quad \Gamma \vdash e2 : t \quad \Gamma \vdash e3 : t}{\Gamma \vdash \text{if } e1 \text{ then } e2 \text{ else } e3 : t} \text{TIf} \\
\\
\frac{\Gamma \vdash e1 : t1 \quad \Gamma, id : t1 \vdash e2 : t2}{\Gamma \vdash \text{let } id = e1 \text{ in } e2 : t2} \text{TLet} \\
\\
\frac{\Gamma \vdash \text{rator} : t1 \rightarrow t2 \quad \Gamma \vdash \text{rand} : t1}{\Gamma \vdash (\text{rator } \text{rand}) : t2} \text{TApp} \\
\\
\frac{\Gamma, id : t1 \vdash e : t2}{\Gamma \vdash \text{proc } (id : t1) \{e\} : t1 \rightarrow t2} \text{TProc}
\end{array}$$

---

**Figure 4.1:** Type Rules for CHECKED

```
type_of_expr : expr -> texpr ea_result
```

reflecting that, given an expression, it returns a function that given a type environment returns either a type or an error. Note, however, that `ea_result` abstracts environments, and not type environments:

```
type 'a ea_result = env -> 'a result
```

We could create a new type constructor, let us call it `tea_result`, where `env` is replaced with `tenv`:

```
type 'a tea_result = tenv -> 'a result
```

But we would also have to duplicate all of `return`, `error`, `>>=`, `lookup_env`, etc. to support this new type and end up having two copies of all these operations (one supporting `ea_result` and one supporting `tea_result`) with the exact same code. Since the only difference between `ea_result` and `tea_result` is the kind of environment they abstract over, we choose to define a more general type constructor `a_result` (“a” for “abstracted”) and have both of these be instances of them:

```
type ('a,'b) a_result = 'b -> 'a result
```

Notice that, contrary to `ea_result` and `tea_result`, the type constructor `a_result` is parameterized over two types,

1. the type `a` representing the result of the computation, and
2. the type `b` representing that over which the function type is being abstracted over.

```

1 type 'a result = Ok of 'a | Error of string
2
3 type ('a,'b) a_result = 'b -> 'a result
4
5 let return : 'a -> ('a,'b) a_result =
6   fun v ->
7     fun env -> Ok v
8
9 let error : string -> ('a,'b) a_result =
10  fun s ->
11    fun env -> Error s
12
13 let (>=>) : ('a,'c) a_result -> ('a -> ('b,'c) a_result) -> ('b,'c) a_result =
14  fun c f ->
15    fun env ->
16      match c env with
17      | Error err -> Error err
18      | Ok v -> f v env
19
20 let (>>+) : ('b,'b) a_result -> ('a,'b) a_result -> ('a,'b) a_result =
21  fun c d ->
22    fun env ->
23      match c env with
24      | Error err -> Error err
25      | Ok newenv -> d newenv

```

reM.ml

**Figure 4.2:** The `a_result` type

The type `('a,'b) a_result`, together with its supporting operations are declared in Fig. 4.2. Notice that the code for the supporting operations, namely `return`, `error`, `>=>` and `>>+` is exactly the same as before. The only difference is their type. That being said, we still call the formal parameter of type `'b` in these operations, `env`.

With the newly declared type constructor `a_result` in place, we can now redefine `ea_result` and `tea_result` as instances of it. Indeed, `ea_result` is simply defined as:

```
type 'a ea_result = ('a,env) a_result
```

and `tea_result` is defined as:

```
type 'a tea_result = ('a,tenv) a_result
```

### Implementation

We next address the implementation of the type-checker for CHECKED. The code is given in Fig. 4.3.

```

1 utop # chk "
2 let add = proc (x:int) { proc (y:int) { x+y }} in (add 1)";;
3 - : texpr ReM.result = Ok (FuncType (IntType, IntType))

```

utop

```

let rec type_of_expr : expr -> texpr tea_result =
2   fun e ->
   match e with
4   | Int _n -> return IntType
   | Var id -> apply_tenv id
6   | IsZero(e) ->
       type_of_expr e >>= fun t ->
8       if t=IntType
       then return BoolType
       else error "isZero: expected argument of type int"
10  | Add(e1,e2) | Sub(e1,e2) | Mul(e1,e2) | Div(e1,e2) ->
       type_of_expr e1 >>= fun t1 ->
       type_of_expr e2 >>= fun t2 ->
12  if (t1=IntType && t2=IntType)
       then return IntType
       else error "arith: arguments must be ints"
14  | ITE(e1,e2,e3) ->
       type_of_expr e1 >>= fun t1 ->
       type_of_expr e2 >>= fun t2 ->
       type_of_expr e3 >>= fun t3 ->
20  if (t1=BoolType && t2=t3)
       then return t2
       else error "ITE: condition not bool/types of then-else do not match"
22  | Let(id,e,body) ->
       type_of_expr e >>= fun t ->
       extend_tenv id t >>+
       type_of_expr body
24  | Proc(var,t1,e) ->
       extend_tenv var t1 >>+
       type_of_expr e >>= fun t2 ->
       return @@ FuncType(t1,t2)
26  | App(e1,e2) ->
       type_of_expr e1 >>=
       pair_of_funcType "app: " >>= fun (t1,t2) ->
       type_of_expr e2 >>= fun t3 ->
30  if t1=t3
       then return t2
       else error "app: type of argument incorrect"
32  | Debug(_e) ->
       string_of_tenv >>= fun str ->
       print_endline str;
       error "Debug: reached breakpoint"
34  | _ -> error "type_of_expr: implement"

```

checker.ml

Figure 4.3: Type checker for CHECKED

### 4.1.4 Adding Letrec

The typing rule for letrec is as follows:

$$\frac{\begin{array}{c} \Gamma, \text{param} : tVar, \text{id} : tPar \rightarrow tRes \vdash \text{body} : tRes \\ \Gamma, \text{id} : tPar \rightarrow tRes \vdash \text{target} : t \end{array}}{\Gamma \vdash \text{letrec id (param:tPar):tRes = body in target} : t} \text{TRec}$$

```

1 let rec type_of_expr : expr -> texpr tea_result = function
2   | Letrec(id,param,tPar,tRes,body,target) ->
3     extend_tenv id (FuncType(tPar,tRes)) >>+
4     (extend_tenv param tPar >>+
5      type_of_expr body >>= fun t ->
6      if t=tRes
7      then type_of_expr target
8      else error "LetRec: Type of rec. function does not match declaration")

```

### 4.1.5 Exercises

**Exercise 4.1.1.** Provide typing derivations for the following expressions:

1. `if zero?(8) then 1 else 2`
2. `if zero?(8) then zero?(0) else zero?(1)`
3. `proc (x:int) { x-2 }`
4. `proc (x:int) { proc (y:bool) { if y then x else x-1 } }`
5. `let x=3 in let y = 4 in x-y`
6. `let two? = proc(x:int) { if zero?(x-2) then 0 else 1 } in (two? 3)`

**Exercise 4.1.2.** Recall that an expression  $e$  is *typable*, if there exists a type environment  $\Gamma$  and a type expression  $t$  such that the typing judgement  $\Gamma \vdash e : t$  is derivable. Argue that the expression  $x\ x$  (a variable applied to itself) is not typable.

**Exercise 4.1.3.** Give a typable term of each of the following types, justifying your result by showing a type derivation for that term.

1. `bool->int`
2. `(bool -> int) -> int`
3. `bool -> (bool -> bool)`
4. `(s -> t) -> (s -> t)`, for any types  $s$  and  $t$ .

**Exercise 4.1.4.** Show that the following term is typable:

```

2 letrec int double (x:int) = if zero?(x)
3   then 0
4   else (double (x-1)) + 2
in double

```

**Exercise 4.1.5.** What is the result of evaluating the following expressions in CHECKED?

```

> (check "
2 letrec int double (x:int) = if zero?(x)
                                then 0
                                else (double (x-1)) + 2
4 in (double 5)")

1 > (check "
letrec int double (x:int) = if zero?(x)
3                                then 0
                                else (double (x-1)) + 2
5 in double")

1 > (check "
letrec bool double (x:int) = if zero?(x)
3                                then 0
                                else -((double -(x,1)), -2)
5 in double")

1 > (check "
letrec bool double (x:int) = if zero?(x)
3                                then 0
                                else 1
5 in double")

1 > (check "
letrec int double (x:bool) = if zero?(x)
3                                then 0
                                else 1
5 in double")

```

**Exercise 4.1.6.** Consider the extension of Exercise 2.2.4 where pairs are added to our language. In order to extend type-checking to pairs we first add pair types to the concrete syntax of types:

```

<Type> ::= int
<Type> ::= bool
<Type> ::= <Type> -> <Type>
<Type> ::= <<Type> * <Type>>
<Type> ::= (<Type>)

```

Recall from Exercise 2.2.4 that expressions are extended with a *pair*(*e*<sub>1</sub>, *e*<sub>2</sub>) construct to build new pairs and an *unpair* (*x*, *y*)=*e*<sub>1</sub> *in* *e*<sub>2</sub> construct that given an expression *e*<sub>1</sub> that evaluates to a pair, binds *x* and *y* to the first and second component of the pair, respectively, in *e*<sub>2</sub>. Here are some examples of expressions in the extended language:

```

pair(3,4)
2 pair(pair(3,4),5)
4 pair(zero?(0),3)
6

```

```
pair(proc (x:int) { x-2 },4)
8
proc (z:<int*int>) { unpair (x,y)=z in x }
10
proc (z:<int*bool>) { unpair (x,y)=z in pair(y,x) }
```

---

*You are asked to give typing rules for each of the two new constructs.*



## Chapter 5

# Modules

### 5.1 Syntax

SIMPLE-MODULES is an extension to the EXPLICIT-REFS language. A program in SIMPLE-MODULES consists of a list of module declarations together with an expression (the “main” expression). Here is an example that consists of one module declaration, the module called `m1`, and a main expression consisting of a `let` expression. A module has an interface and a body.

```
1 module m1
  interface
3   [a : int
    b : int
5   c : int]
  body
7   [a = 33
    x = a-1 (* =32 *)
9   b = a-x (* = 1 *)
    c = x-b] (* =31 *)
11 let a = 10
    in ((from m1 take a) - (from m1 take b))-a
```

### 5.1.1 Concrete Syntax

```

<Program>      ::=  {<ModuleDefn>}* <Expression>

<ModuleDefn>   ::=  module <Identifier> interface <Iface> body <ModuleBody>

<Iface>        ::=  [{<Decl>}*]
<Decl>         ::=  <Identifier> : <Type>

<ModuleBody>   ::=  [{<Defn>}*]
<Defn>         ::=  <Identifier> = <Expression>

<Expression>   ::=  ...asbefore...
<Expression>   ::=  from<Identifier> take<Identifier>

```

### 5.1.2 Abstract Syntax

```

type expr =
2   ...
  | QualVar of string*string
4 and
  texpr =
6   | IntType
  | BoolType
8   | UnitType
  | FuncType of texpr*texpr
10  | RefType of texpr
and
12  vdecl = string*texpr
and
14  vdef = string*expr

16 type interface = ASimpleInterface of vdecl list
type module_body = AModBody of vdef list
18 type module_decl = AModDecl of string*interface*module_body
type prog = AProg of (module_decl list)*expr

```

## 5.2 Interpreter

### 5.2.1 Specification

$$\begin{array}{c}
\frac{\text{mdecls}, \epsilon \Downarrow \rho \quad \text{e}, \rho \Downarrow v}{\text{AProg}(\text{mdecls}, \text{e}) \Downarrow v} \text{EProg} \\
\\
\frac{}{\epsilon, \rho \Downarrow \rho} \text{EMDeclsEmpty} \\
\\
\frac{\text{body}, \rho \Downarrow \sigma \quad \text{ms}, \rho \oplus \{\text{id} := \sigma\} \Downarrow \tau}{\text{AModDecl}(\text{id}, \text{iface}, \text{body}) \text{ ms}, \rho \Downarrow \tau} \text{EMDeclsCons} \\
\\
\frac{}{\epsilon, \rho \Downarrow \rho} \text{EBValsEmpty} \\
\\
\frac{\text{e}, \rho \Downarrow v \quad \text{vs}, \rho \oplus \{\text{id} := v\} \Downarrow \tau}{(\text{id}, \text{e}) \text{ vs}, \rho \Downarrow \tau} \text{EBValsCons} \\
\\
\frac{\rho(\text{mid}) = \sigma \quad \sigma(\text{vid}) = v}{\text{QualVar}(\text{mid}, \text{vid}), \rho \Downarrow v} \text{EQualVar}
\end{array}$$

### 5.2.2 Implementation

We first extend environments to support bindings for modules:

```

1 type env =
  | EmptyEnv
3   | ExtendEnv of string*exp_val*env
  | ExtendEnvRec of string*string*Ast.expr*env
5   | ExtendEnvMod of string*env*env

```

ds.ml

Evaluation of programs consists in first evaluating all module definitions producing an environment as a result, and then evaluating the main expression using this environment. The former is achieved with the function `eval_module_definitions : module_decl list -> env ea_result`:

```

1 let eval_prog (AProg(ms,e)) : exp_val ea_result =
2   eval_module_definitions ms >>+
   eval_expr e

```

```

1 let rec eval_expr : expr -> exp_val ea_result =
   fun e ->
3   match e with
  | Int(n) ->
5     return @@ NumVal n
  | Var(id) ->
7     apply_env id
  | QualVar(module_id,var_id) ->
9     apply_env_qual module_id var_id

```

```

...
11 and
    eval_module_definition : module_body -> env ea_result =
13   fun (AModBody vdefs) ->
    lookup_env >>= fun glo_env ->
15   (List.fold_left (fun loc_env (var,decl) ->
        loc_env >>+
17     (append_env_rev glo_env >>+
        eval_expr decl >>=
19     extend_env var))
    (empty_env ()))
21   vdefs)

23 and
    eval_module_definitions : module_decl list -> env ea_result =
    fun ms ->
25   List.fold_left
    (fun curr_en (AModDecl(mname,minterface,mbody)) ->
27     curr_en >>+
        (eval_module_definition mbody >>=
29     extend_env_mod mname))
    lookup_env
31   ms

33 and
    eval_prog (AProg(ms,e)) : exp_val ea_result =
    eval_module_definitions ms >>+
35   eval_expr e

```

interp.ml

```

1 utop # interp "
    module m1 interface
3   [u : int] body
    [u = 44]

5   module m2 interface
    [v : int] body
7   [v = (from m1 take u)-11]
    let a=zero?(0)
9   in debug(0)";;
    Environment:
11   (a, BoolVal true)
    Module m2[(v, NumVal 33)]
13   Module m1[(u, NumVal 44)]
    Store:
15   Empty
- : Ds.exp_val ReM.result = ReM.Ok Ds.UnitVal

```

utop

## 5.3 Type-Checking

### 5.3.1 Specification

|                                                   |                                         |
|---------------------------------------------------|-----------------------------------------|
| Judgements for typing programs                    | $\vdash \text{AProg}(\text{ms}, e) : t$ |
| Judgements for typing expressions                 | $\Delta; \Gamma \vdash e : t$           |
| Judgements for typing list of module declarations | $\Delta_1 \vdash \text{ms} : \Delta_2$  |

$$\begin{array}{c}
\frac{\epsilon \vdash \text{ms} :: \Delta \quad \Delta; \epsilon \vdash e :: t}{\vdash \text{AProg}(\text{ms}, e) :: t} \text{TProg} \\
\\
\frac{m \in \text{dom}(\Delta) \quad x \in \text{dom}(\Delta(m)) \quad \Delta(m, x) = t}{\Delta; \Gamma \vdash \text{from } m \text{ take } x :: t} \text{TFromTake} \\
\\
\frac{}{\Delta \vdash \epsilon :: \epsilon} \text{TModE} \\
\\
\frac{
\begin{array}{c}
[x_i]_{i \in I} \triangleleft [y_j]_{j \in J} \\
(\Delta_1; [y_1 := s_1] \dots [y_{j-1} := s_{j-1}] \Gamma \vdash e_j :: s_j)_{j \in J} \\
(t_i = s_{f(i)})_{i \in I} \\
m[x_i : t_i]_{i \in I} \Delta_1 \vdash \text{ms} :: \Delta_2
\end{array}
}{\Delta_1 \vdash m[x_i : t_i]_{i \in I} [y_j = e_j]_{j \in J} \text{ms} :: m[x_i : t_i]_{i \in I} \Delta_2} \text{TModNE}
\end{array}$$

Figure 5.1: Typing rules for SIMPLE-MODULES

$\Gamma$  is the standard type environment from before  $\Delta$  is a module type environment and is required for typing the expression `from m take x`

A **module type** is an expression of the form  $m[u_1 : t_1, \dots, u_n : t_n]$ . A **module type environment** is a sequence of module types.

$$\Delta ::= \epsilon \mid m[u_1 : t_1, \dots, u_n : t_n] \Delta$$

We use letters  $\Delta$  to denote module type environments. The empty module type environment is written  $\epsilon$ . If  $m[u_1 : t_1, \dots, u_n : t_n] \in \Delta$ , then we  $m \in \text{dom}(\Delta)$ . Moreover, in that case, have  $u_i \in \text{dom}(\Delta(m))$ , for  $i \in 1..n$ , and also  $\Delta(m, u_i) = t_i$ .

There is just one typing rule for typing programs, namely TProg. There is one new typing rule for expressions, namely TFromTake, it allows to type qualified variables. There are two typing rules for lists of module definitions: one for when the list is empty (TModE) and one for when it is not (TModNE). Regarding the latter,

- $\Delta_2$  is the type of the list of modules `ms`
- $m[x_i : t_i]_{i \in I} \Delta_1$  is the type of the list of modules that `ms` can use
- $[x_i]_{i \in I} \triangleleft [y_j]_{j \in J}$  means that the list of variables  $[x_i]_{i \in I}$  is a sublist of the list of variables  $[y_j]_{j \in J}$ . This relation determines an injective, order preserving function  $f : I \rightarrow J$

### 5.3.2 Implementation

```

type tenv =
2 | EmptyTEnv
  | ExtendTEnv of string*texpr*tenv
4 | ExtendTEnvMod of string*tenv*tenv

```

dst.ml

```

let rec
2   type_of_prog (AProg (ms,e)) =
   type_of_modules ms >>+
4   type_of_expr e
and
6   type_of_modules : module_decl list -> tenv tea_result =
   fun mdecls ->
8   List.fold_left
     (fun curr_tenv (AModDecl(mname,ASimpleInterface(expected_iface),mbody)) ->
10      curr_tenv >>+
        (type_of_module_body mbody >>= fun i_body ->
12          if (is_subtype i_body expected_iface)
              then
14              extend_tenv_mod mname (var_decls_to_tenv expected_iface)
              else
16              error @@ "Subtype failure: "^mname)
        )
18      lookup_tenv
        mdecls
20 and
   type_of_module_body : module_body -> tenv tea_result =
22   fun (AModBody vdefs) ->
     lookup_tenv >>= fun glo_tenv ->
24     (List.fold_left (fun loc_tenv (var,decl) ->
        loc_tenv >>+
26         (append_tenv_rev glo_tenv >>+
            type_of_expr decl >>=
28             extend_tenv var))
        (empty_tenv ()))
30     vdefs) >>= fun tmbody ->
     return @@ reverse_tenv tmbody
32 and
   type_of_expr : expr -> texpr tea_result =
34   fun e ->
     match e with
36     | Int n -> return IntType
     | Var id -> apply_tenv id
38     | QualVar(module_id,var_id) ->
        apply_tenv_qual module_id var_id
40     ...

```

## 5.4 Further Reading

Module inclusion Private types First-class modules

# Appendix A

## Supporting Files

### A.1 File Structure

Typical file structure for an interpreter (in this example, ARITH).

```
.
|____arith.opam
|____test
| |____dune
| |____test.ml
|____dune-project
|____build
|____.ocamlinit
|____src
| |____ds.ml
| |____ast.ml
| |____.merlin
| |____lexer.mll
| |____dune
| |____interp.ml
| |____parser.mly
```

The source files are in the `src` directory and the unit tests are in the `test` directory.

|                         |                                                                                 |
|-------------------------|---------------------------------------------------------------------------------|
| <code>ast.ml</code>     | Abstract Syntax                                                                 |
| <code>ds.ml</code>      | Supporting data structures including expressed values, environments and results |
| <code>interp.ml</code>  | Interpreter                                                                     |
| <code>lexer.mll</code>  | Lexer generator                                                                 |
| <code>parser.mly</code> | Parser generator                                                                |
| <code>test.ml</code>    | Unit tests                                                                      |
| <code>.ocamlinit</code> | Loaded by utop upon execution; opens some modules                               |

We use the dune build system for OCaml. You can find documentation on dune at <https://readthedocs.org/projects/dune/downloads/pdf/latest/>. Some common dune commands:

- Builds the project and then runs utop

```
$ dune utop
```

- Builds the project

```
$ dune build
```

- Clean the current project (erasing `_build` directory)

```
$ dune clean
```

- Run tests (building if necessary)

```
$ dune runtest
```



## Appendix B

# Solution to Selected Exercises

### Section 2.1

**Answer B.0.1** (Exercise 2.2.1). *Sample expressions of each of the following types are:*

1. *expr*. An example is: `Int 2`.
2. *env*. Examples are: `EmptyEnv` and `ExtendEnv("x", NumVal 2, EmptyEnv)`
3. *exp\_val*. Examples are: `NumVal 3` and `BoolVal true`.
4. *exp\_val result*. Examples are: `Ok (NumVal 3)` and `Ok (BoolVal true)` and `Error "oops"`.
5. *int result*. Examples are: `Ok 1` and `Error "oops"`.
6. *env result*. Examples are: `Ok EmptyEnv` and `Ok (ExtendEnv("x", NumVal 2, EmptyEnv))`.
7. *int ea\_result*. Examples are: `return 2` and `error "oops"`.
8. *exp\_val ea\_result*. Examples are: `return (NumVal 2)` and `return (BoolVal true)` and `error "oops"`.  
Also, `apply_env "x"`.
9. *env ea\_result*. Examples are: `return (EmptyEnv)` and `error "oops"`. Also, `extend_env "x" (NumVal 7)`.

### Section 2.3

**Answer B.0.2** (Exercise 2.4.7).

```
1 let even = proc (e) { proc (o) { proc (x) {  
2   if zero?(x)  
3   then zero?(0)  
4   else (((o e) o) (x-1)) }}}  
5 in  
6 let odd = proc(e) { proc (o) { proc (x) {  
7   if zero?(x)  
8   then zero?(1)  
9   else (((e e) o) (x-1)) }}}  
10 in (((even even) odd) 4)
```

**Answer B.0.3** (Exercise 2.4.8).

```

2  let f =
    proc (g) {
      proc (leaf) {
4     proc (depth) {
        if zero?(depth) then (leaf, leaf)
6         else (((g g) leaf) (depth-1)), (((g g) leaf) (depth-1))) }}}
    in let pbt = proc (leaf) { proc (height) { ((f f) leaf) height) }}
8    in ((pbt 2) 3)

```

**Section 2.5****Answer B.0.4** (Exercise 2.5.2).

```

2  # interp "
    let l1 = cons(1, cons(2, cons(3, emptylist)))
    in let l2 = cons(4, cons(5, emptylist))
4    in letrec append(l1) = proc (l2) {
        if empty?(l1)
6         then l2
        else cons(hd(l1), ((append tl(l1)) l2))
8         }
    in ((append l1) l2)
10  ";
    - : exp_val Rec.Ds.result =
12  Ok (ListVal [NumVal 1; NumVal 2; NumVal 3; NumVal 4; NumVal 5])

14
16  # interp "
    let l = cons(1, cons(2, cons(3, emptylist)))
    in let succ = proc (x) { x+1 }
18    in letrec map(l) = proc (f) {
        if empty?(l)
20         then emptylist
        else cons((f hd(l)), ((map tl(l)) f))
22         }
    in ((map l) succ)
24  ";
    - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 2; NumVal 3; NumVal
26  4])

28
30  # interp "
    let l = cons(1, cons(2, cons(1, emptylist)))
    in let is_one = proc (x) { zero?(x-1) }
32    in letrec filter(l) = proc (p) {
        if empty?(l)
34         then emptylist
        else (if (p hd(l))
36              then cons(hd(l), ((filter tl(l)) p))
              else ((filter tl(l)) p))
38         }
    in ((filter l) is_one)";
40  - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 1; NumVal 1])

```

**Answer B.0.5** (Exercise 2.5.3). 1. *debug* Must be placed in the *then* case:

```

let z = 0
2 in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then debug(1) else ((prod n) (f (n-1)))
4 in (f 10)

```

2. Two possible solutions are:

```

let z = 0
2 in debug(let prod = proc (x) { proc (y) { x*y }})
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10))

```

or:

```

let z = 0
2 in let prod = debug(proc (x) { proc (y) { x*y }})
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10)

```

3. *debug* must be placed in the body of *proc* (x):

```

let z = 0
2 in let prod = proc (x) { debug(proc (y) { x*y })}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10)

```

---

---

---

---

---

---

---

---